

Figure 1. Fields near the boundary between two media illustrating the geometry and notation employed in the text to discuss the boundary conditions on (a) **H**, and (b) **B**

equations (7) and (3). Integration of (7) yields

$$\oint \mathbf{H} \cdot \mathrm{d}\mathbf{l} = 0, \tag{11}$$

and evaluating the line integral around a rectangular contour intersecting the boundary between two regions, in which the magnetic fields are $\mathbf{H}_1$ and $\mathbf{H}_2$, as shown in Figure 1(a),

$$\mathbf{H}_1 \cdot \mathrm{d}\mathbf{l} - \mathbf{H}_2 \cdot \mathrm{d}\mathbf{l} + 0(\mathrm{d}h) = 0, \tag{12}$$

(assuming that dh can be made arbitrarily small). Equation (12) implies that the tangential components of **H** are continuous across the boundary, and can also be expressed as

$$\hat{\mathbf{n}} \times (\mathbf{H}_1 - \mathbf{H}_2) = 0, \tag{13}$$

where $\hat{\mathbf{n}}$ is a unit vector normal to the surface. In terms of the scalar potentials,

$$\hat{\mathbf{n}} \times (\nabla\phi_1 - \nabla\phi_2) = 0 \tag{14}$$

and, integrating along the boundary yields, in many circumstances,

$$\phi_1 = \phi_2. \tag{15}$$

Thus the potential is continuous across the boundary.

The second boundary condition is derived from $\operatorname{div} \mathbf{B} = 0$ by applying Gauss's theorem to yield

$$\int_S \mathbf{B} \cdot d\mathbf{S} = 0. \tag{16}$$

For a small cylindrical volume intersecting the boundary, indicated by the dotted lines in Figure 1(b), the surface integral yields

$$\mathbf{B}_2 \cdot \hat{\mathbf{n}}\, dS - \mathbf{B}_1 \cdot \hat{\mathbf{n}}\, dS + 0(dh) = 0, \tag{17}$$

indicating that the normal component of $\mathbf{B}$ is continuous. If the two regions have magnetizations $\mathbf{M}_1$ and $\mathbf{M}_2$, substitution of $\mathbf{B} = \mu_0(-\nabla\phi + \mathbf{M})$ yields

$$(-\nabla\phi_1 + \mathbf{M}_1) \cdot \hat{\mathbf{n}} = (-\nabla\phi_2 + \mathbf{M}_2) \cdot \hat{\mathbf{n}}. \tag{18}$$

This is the required boundary condition on the gradient of ϕ. It can be seen from equation (18) that, by analogy with electrostatics, the term $\mathbf{M} \cdot \hat{\mathbf{n}}$ plays the role of a surface magnetic charge (pole) density.

The normal component of the magnetic field $\mathbf{H}$ has a discontinuity equal to the difference of the components of the magnetizations. Consequently the magnetic field inside the magnet opposes the magnetization and is known as the demagnetizing field.

4. THE MODEL PROBLEM

The computational problem is to determine the magnetic field in the regions inside and outside a two-dimensional rectangular magnet by solving Poisson's equation (10) for the scalar potential. It is assumed that the magnet is uniformly magnetized, so that $\mathbf{M}(\mathbf{r})$ is a constant vector, in which case equation (10) reduces to Laplace's equation,

$$\frac{\partial^2\phi}{\partial x^2} + \frac{\partial^2\phi}{\partial y^2} = 0, \tag{19}$$

everywhere except on the boundary of the magnet. On the latter, the potential is continuous, but the components of its gradient normal to the surface change by the normal component of the magnetization, in accordance with equation (18).

At large distances from the magnet the potential will resemble that of a small magnetic dipole of moment $\mathbf{m} = \mathbf{M}V$, where V is the volume of the



Figure 2. Schematic diagram showing the positions of the magnet and the outer boundary, and indicating the directions of the magnetic field. Because of symmetry it is necessary to consider only the region in the positive quadrant

magnet, i.e.

$$\phi(\mathbf{r}) = -\frac{1}{4\pi}\mathbf{m}\cdot \mathrm{grad}\left(\frac{1}{r}\right). \tag{20}$$

In principle it is possible to match the numerical solutions of (19) to this expression on a distant rectangular boundary, but to avoid this added complication it will be assumed here that ϕ is essentially zero on that boundary, as shown in Figure 2. The effect of this approximation on the final solutions can be investigated by increasing the size of the large rectangle.

5. THE FINITE DIFFERENCE EQUATIONS

Laplace's equation (19) can be solved by approximating the derivatives of ϕ by finite difference formulae. The second derivative of a function of a single variable $f(x)$ tabulated at equal intervals of x can be approximated,[2] using Taylor's expansion, by

$$\frac{\mathrm{d}^2 f}{\mathrm{d}x^2} \approx \frac{1}{h^2}\{f(x+h)+f(x-h)-2f(x)\}, \tag{21}$$

where h is the interval. For a function of two variables, $f(x, y)$ can be specified at points on a square mesh labelled by integers i and j, so that $x = ih$; $y = jh$ $(i, j = 1, 2, 3, \ldots)$.

Equation (21) enables Laplace's equation to be replaced by a set of finite element equations:

$$\frac{1}{h^2}\{\phi_{i+1,j}+\phi_{i-1,j}+\phi_{i,j+1}+\phi_{i,j-1}-4\phi_{i,j}\}=0, \tag{22}$$

for each point (i, j). Near the boundaries of the region this equation must be modified in an appropriate way, described in detail below, to take into account the physical boundary conditions. The resulting set of equations, one equation for each mesh point, can be solved for the $\phi_{i,j}$ either by direct matrix methods or by an iterative process (see, for example, ref. 3). For large matrices the second method has the advantage that the zeros are preserved throughout, and consequently less computer storage is required. An iterative approach is employed here.

The five values of ϕ in equation (22) are said to form a star (Figure 3). If four of the values are known approximately, equation (22) can be employed to determine an improved value for the fifth. In an iterative process initial values of ϕ, $\phi_{i,j}^{(0)}$ say, must be assigned to each mesh point. Generally, it is not essential that these initial values are a close approximation to the final solution. On the boundaries the ϕ-values may be known exactly from the outset, but often the $\phi_{i,j}^{(0)}$-values inside the region can only be chosen somewhat arbitrarily. Frequently the $\phi_{i,j}^{(0)}$ are set equal to a constant value



Figure 3. The star of function values required to approximate $\nabla^2\phi$ at a general mesh point (i, j). The mesh is assumed to be square and the mesh size is h

(0.5 in the program presented here). The initial $\phi_{i,j}^{(0)}$ can be improved by applying equation (22) to each mesh point in turn, giving quantities $\phi_{i,j}^{(1)}$. Explicitly,

$$\phi_{i,j}^{(1)} = \tfrac{1}{4}\{\phi_{i+1,j}^{(0)} + \phi_{i-1,j}^{(0)} + \phi_{i,j-1}^{(0)} + \phi_{i,j+1}^{(0)}\}, \tag{23}$$

If the new $\phi_{i,j}^{(1)}$ are substituted into equation (22) the bracket on the left-hand side will not be exactly zero, but will have a residual value, R_{ij} say, which is some measure of the discrepancy between $\phi_{i,j}^{(1)}$ and the true solution ϕ. Repeating the procedure, new values $\phi_{i,j}^{(2)}$ can be computed from the $\phi_{i,j}^{(1)}$ using a formula similar to (23). This iterative process is continued until the ϕ values do not alter, within a specified accuracy, between one scan of the mesh points and the next.

Iterative formulae like (23) employing only the old $\phi^{(0)}$ values on the right-hand side are said to be of Jacobi type. When performing the calculations with a computer it is more natural to use the newly calculated ϕ values in the right-hand side of (23) as soon as possible. Thus, for example, if the mesh is scanned column by column the quantities $\phi_{i-1,j}^{(1)}$ and $\phi_{i,j-1}^{(1)}$ will have been computed before $\phi_{i,j}^{(1)}$ is evaluated, and equation (23) can be replaced by

$$\phi_{i,j}^{(1)} = \tfrac{1}{4}\{\phi_{i+1,j}^{(0)} + \phi_{i-1,j}^{(1)} + \phi_{i,j-1}^{(1)} + \phi_{i,j+1}^{(0)}\}. \tag{24}$$

This expression gives rise to a Gauss–Seidel scheme. It can be shown[3] that this iterative process converges more rapidly than the simpler Jacobi method.

More complicated iteration formulae than (24) have been devised which give even more rapid convergence than the Gauss–Seidel method. One such procedure,[3] which changes the old field value by adding to it a small fraction of the old residual R_{ij} is known (amongst other names) as successive over-relaxation (SOR). For the nth iteration, assuming column-by-column scanning of the mesh, the appropriate formula is

$$\phi_{i,j}^{(n)} = \phi_{i,j}^{(n-1)} + \frac{\alpha}{4}\{\phi_{i+1,j}^{(n-1)} + \phi_{i-1,j}^{(n)} + \phi_{i,j-1}^{(n)} + \phi_{i,j+1}^{(n-1)} - 4\phi_{i,j}^{(n-1)}\}, \tag{25}$$

where α is a parameter which usually lies between 1 and 2 in practice.[3] This method has been employed in the present work.

Because of the symmetry of the problem it is necessary to consider only one-quarter of the total region, for example the first quadrant shown in Figure 4. The iteration formulae (23)–(25) cannot be employed for points on the surface of the magnet because Laplace's equation is not valid there. Nor can these formulae be used as they stand for points on the boundaries of the region, because some of the ϕ-values in the star formulae for the boundary points will be outside the region. The latter situation can be illustrated by



Figure 4. A typical mesh covering the region of interest. The area $OACB$ represents the portion of the magnet in this quadrant

considering the boundary OY, for which $i=1$, giving for the residuals R_{1j}

$$R_{1j} = \phi_{2,j} + \phi_{0,j} + \phi_{1,j+1} + \phi_{1,j-1} - 4\phi_{1,j}. \tag{26}$$

The values $\phi_{0,j}$ lying outside the field region are sometimes called 'fictitious' values, and are often labelled by an asterisk added as a superscript, e.g. $\phi^*_{0,j}$. Because of the special conditions applying at the boundaries the fictitious values required to calculate the residuals can usually be expressed as functions of the $\phi_{i,j}$ values inside the region. For the present problem the appropriate special finite difference formulae required for the region boundaries and the magnet surface are now derived in detail.

5.1 The outer boundary

All $\phi_{i,j}=0$, and this boundary is excluded from the iterative process.

5.2 The symmetry axis OY

The line OY lies along the y-coordinate axis (Figure 4). The field in the negative x region ($y>0$) is the mirror image of that in the first quadrant, i.e.

$$\phi(-x, y)=\phi(x, y).$$

Hence the fictitious field values $\phi^*_{0,j}$ (see Figure 5a) are given by

$$\phi^*_{0,j}=\phi_{2,j}. \tag{27}$$

Employing this in the star formula for R_{ij} gives a SOR formula:

$$\phi^{(n)}_{1,j}=\phi^{(n-1)}_{1,j}+\frac{\alpha}{4}\{2\phi^{(n')}_{2,j}+\phi^{(n')}_{1,j-1}+\phi^{(n')}_{1,j+1}-4\phi^{(n-1)}_{1,j}\}, \tag{28}$$



Figure 5. Function values involved in treating typical points on the boundaries: (a) symmetry line OY showing the position of the fictitious point $\phi^*_{0,j}$; (b) symmetry line OX showing the position of the fictitious point $\phi^*_{i,0}$; (c) magnet boundary AC; (d) magnet boundary BC

(c)



(d)



Figure 5c and d

where n' is n or $(n-1)$ and indicates that the most recently calculated ϕ-values are to be inserted. This formula applies for $j>1$, and the origin O and point B are excluded as they also lie on other symmetry axes.

5.3 The symmetry axis *OX*

Line OX lies along the x-axis. From the symmetry, the magnetic field **H** is everywhere parallel to OY, so that

$$H^y = -\frac{\partial\phi}{\partial y} = H, \tag{29}$$

$$H^x = -\frac{\partial\phi}{\partial x} = 0, \tag{30}$$

The second equation means that all the ϕ-values on OX are equal. Furthermore, the continuity of **H** across OX requires that the gradient $\partial\phi/\partial y$ should be continuous. Defining fictitious values $\phi^*_{i,0}$ as in Figure 5(b), and employing a simple forward difference formula for the gradient, this condition gives

$$\frac{1}{h}\{\phi_{i,2}-\phi_{i,1}\}=\frac{1}{h}\{\phi_{i,1}-\phi^*_{i,0}\},$$

or

$$\phi^*_{i,0}=2\phi_{i,1}-\phi_{i,2}. \tag{31}$$

Substitution into the star formula for the residual, the SOR iteration formula becomes

$$\phi^{(n)}_{i,1}=\phi^{(n-1)}_{i,1}+\frac{\alpha}{4}\{\phi^{(n')}_{i+1,1}+\phi^{(n')}_{i-1,1}-2\phi^{(n-1)}_{i,1}\}, \tag{32}$$

where n' is n or $(n-1)$ depending on how the boundary mesh points are scanned. In fact this equation is redundant because the ϕ are all equal by equation (30). Hence, since ϕ is zero on the outer boundary, it is necessary to set

$$\phi=0,$$

at all points on the line, including the origin O.

5.4 The magnet boundary AC

The condition on tangential **H** requires that ϕ is continuous on the boundary, and this is automatically ensured in the iteration. Since the magnetization has no component perpendicular to AC, the continuity of the normal component of **B** implies that $\partial\phi/\partial x$ is continuous. Employing a forward difference formula, see Figure 5c,

$$\frac{1}{h}\{\phi_{i+1,j}-\phi_{i,j}\}=\frac{1}{h}\{\phi_{i,j}-\phi_{i-1,j}\},$$

or

$$\phi_{i,j}=\tfrac{1}{2}\{\phi_{i+1,j}+\phi_{i-1,j}\}. \tag{33}$$

Hence on this boundary, excluding point C, which requires a different treatment because it also lies on boundary BC,

$$\phi^{(n)}_{i,j}=\phi^{(n-1)}_{i,j}+\frac{\alpha}{2}\{\phi^{(n')}_{i+1,j}+\phi^{(n')}_{i-1,j}-2\phi^{(n-1)}_{i,j}\}, \tag{34}$$

where again n' is either n or $(n-1)$ as appropriate.

5.5 The magnet boundary *BC*

Again ϕ is continuous, but now, from equation (18), the gradient in y has a discontinuity equal to $|\mathbf{M}|$, i.e.

$$-\frac{\partial\phi}{\partial y}\bigg|_{\text{in}}+M=-\frac{\partial\phi}{\partial y}\bigg|_{\text{out}}. \tag{35}$$

In terms of finite differences, this becomes (see Figure 5d):

$$-\frac{1}{h}\{\phi_{i,j+1}-\phi_{i,j}\}=-\frac{1}{h}\{\phi_{i,j}-\phi_{i,j-1}\}+M,$$

or

$$\phi_{i,j}=\tfrac{1}{2}\{\phi_{i,j+1}+\phi_{i,j-1}+Mh\}. \tag{36}$$

Therefore, excluding point *C*, but including point *B*,

$$\phi_{i,j}^{(n)}=\phi_{i,j}^{(n-1)}+\frac{\alpha}{2}\{\phi_{i,j+1}^{(n')}+\phi_{i,j-1}^{(n')}+Mh-2\phi_{i,j}^{(n-1)}\}. \tag{37}$$

5.6 The point *C*

This point is difficult to treat satisfactorily. Only a very approximate expression for ϕ will be used here, it being assumed that a simple average of the expressions for *AC* and *BC*, given by (33) and (36), is appropriate. Hence, adding,

$$2\phi_{i,j}=\tfrac{1}{2}\{\phi_{i+1,j}+\phi_{i-1,j}\}+\tfrac{1}{2}\{\phi_{i,j+1}+\phi_{i,j-1}+Mh\}, \tag{38}$$

and

$$\phi_{i,j}^{(n)}=\phi_{i,j}^{(n-1)}+\frac{\alpha}{4}\{\phi_{i+1,j}^{(n')}+\phi_{i,j+1}^{(n')}+\phi_{i,j+1)}^{(n')}+\phi_{i,j-1}^{(n')}+Mh-4\phi_{i,j}^{(n-1)}\}. \tag{39}$$

6. THE COMPUTER PROGRAM

The iteration scheme is readily programmed for a computer, and a suitable FORTRAN coding is given at the end of the chapter. The flow diagram Figure 6 indicates a few practical details. The iteration loop for the $\phi_{i,j}$ is included in the main routine. Subroutines calculate the magnetic field and handle the output of data.

It is necessary to keep a running count of the number of iterations performed, so that the process can be terminated if it shows no sign of converging to the required accuracy. There are several criteria for satisfactory convergence that can be employed, and in the appended program two different criteria are used in series. In the first, the residual with the largest



Figure 6. Flow diagram for the computer program

magnitude is determined for the nth iteration, and if this is less than some prescribed value, R_m say, the process is terminated. In some cases this criterion might be too strong, for it is conceivable that all but a few residuals are much smaller than R_m, and the solution for the $\phi_{i,j}$ might be acceptable at that stage. Therefore the second criterion used is that the root-mean-square average residual over the total number of scanned mesh points N is less than some small value R_{ms}, i.e.

$$\left(\frac{1}{N}\sum_{i,j} R_{i,j}^2\right)^{\frac{1}{2}} < R_{ms}. \tag{40}$$

The iteration is terminated when one of these two criteria is first satisfied.

The program produces, on a line printer or a VDU, a mesh pattern showing the position and size of the magnet. It also outputs the potential ϕ_{ij} and magnetic field magnitudes and directions in array form.

The components of the magnetic field $\mathbf{H}$ are calculated by subroutine THEMA using the simple finite difference formulae for the gradients:

$$H_{i,j}^x = -\frac{1}{h}(\phi_{i,j} - \phi_{i-1,j}); \qquad H_{i,j}^y = -\frac{1}{h}(\phi_{i,j} - \phi_{i,j-1}). \tag{41}$$

The lines OX and OY are not included, but from symmetry $H^x = 0$ in these directions. It may be possible, depending on the facilities available, to use graph plotting facilities to display the vector field $\mathbf{H}$ directly. However, the present program outputs arrays containing the directions $\theta_{i,j}$ (in degrees with line OX corresponding to $\theta = 0°$) and magnitudes $H_{i,j}$ of the field. These are calculated from

$$\theta_{i,j} = \tan^{-1}(H^y/H^x); \qquad H_{i,j} = [(H^x)^2 + (H^y)^2]^{\frac{1}{2}}, \tag{42}$$

care being taken to adjust the arc tangent as appropriate, depending on the signs of H^x and H^y. To illustrate the use of these formulae, consider the calculation of $\mathbf{H}$ at the point $i = 5$, $j = 5$ from some typical values of the potential, noting that $h = 0.1$, i.e.

$$\begin{aligned} H^x &= -\frac{1}{h}\{\phi_{5,5} - \phi_{4,5}\} = -10.0\{0.38 - 0.65\} = 2.7, \\ H^y &= -\frac{1}{h}\{\phi_{5,5} - \phi_{5,4}\} = -10.0\{0.38 - 0.30\} = -0.8. \end{aligned} \tag{43}$$

Hence expression (42) gives

$$\begin{aligned} H_{5,5} &= \{(2.7)^2 + (0.8)^2\}^{\frac{1}{2}} = 2.82, \\ \theta_{5,5} &= \tan^{-1}(-0.8/2.7) = -16° \equiv +344°. \end{aligned} \tag{44}$$

These answers can be located in the typical output given in this chapter.



Figure 7. A typical magnetic field distribution in the positive quadrant plotted from results obtained from the computer program

The values of $H_{i,j}$ and $\theta_{i,j}$ enable the vector magnetic field to be plotted at each mesh point. For the mesh point (5, 5) for example, the vector $\mathbf{H}$ can be represented by a line of length 2.82 units (on some suitable scale) drawn to make an angle of +344° with the OX-axis. A plot of all the field vectors corresponding to the typical output is shown in Figure 7.

7. RUNNING THE PROGRAM

The small list of input parameters is as follows:

IN, JN	Number of mesh points in the x- and y-directions, respectively.
IM, JM	Number of mesh points in the x- and y-directions occupied by the magnet.
H	Mesh size (metres) (written as h in the text).
ALPHA	Convergence parameter ($1<\alpha<2$).
AMAG	The magnetization (Am^{-1}).

The basic equations of section 2 are appropriate when the physical quantities are measured in SI units. In this system the magnetic induction vector **B** has units of tesla (T), and both the magnetization **M** and magnetic field **H** have units of ampere metre^{-1} (Am^{-1}). The values of $|\mathbf{M}|$ for typical permanent magnets are in the range 10^5–10^6 Am^{-1}. Spatial dimensions are measured in metres. Since M and h enter the potential equations only through the product Mh, the solution for a given value of Mh is applicable to many problems with different combinations of these parameters. However, it should be noted that the magnitude of the magnetic field **H** depends on the value of h in each case, through the gradient, and further, the magnitude of **H** scales linearly with M.

A typical set of input parameters is as follows:

9	10	4	5	0.1	1.5	5.0
(IN)	(JN)	(IM)	(JM)	(H)	(ALPHA)	(AMAG)

The first six parameters are read in on one line using FORMAT (4I4, 2F5.2) and AMAG is read separately using FORMAT (F5.2). It should be noted that the same device number has been employed for both input and output. The values of the input data are printed out immediately followed by a mesh pattern showing the position of the magnet.

The next three numbers printed give information on the convergence of the iteration process. The parameters determining the conditions for termination of the iteration have been preset in the program. The largest residual, REMAX, and the root mean square of the residuals, RESUM, are determined after each scan of the mesh. These are compared with two parameters, EPMAX and EPSUM respectively, which have both been given values of 10^{-4}. If REMAX is less than EPMAX, or if RESUM is less than EPSUM, it is assumed that the solution is sufficiently accurate for the present purpose. From the typical results given here it is seen that iteration has terminated when RESUM ($=0.95\times10^{-4}$) became smaller than 10^{-4}. The largest residual at that time was reasonably close to 10^{-4} (REMAX$=2.4\times10^{-4}$), giving confidence in the solution obtained. The number of iterations needed is also printed out, being 28 in the example given. A control in the program stops the calculations if the integer NCOUN, counting the number of iterations, exceeds 100. It is a simple matter to change the preset parameters by making minor modifications to the program coding.

Finally, the arrays containing $\phi_{i,j}$, $\theta_{i,j}$ and $H_{i,j}$ are printed out. To understand the solution for the vector field **H**, the latter should be plotted in a form like Figure 7. It can be seen from that figure that inside the magnet the field is almost parallel to the direction of magnetization, but is in the opposite direction. This is the demagnetizing field. Outside the magnet the field is strongest near the magnet's face BC, where it is roughly parallel to the direction of **M**. This behaviour is that expected from the simplest model of a bar magnet in which (fictitious) magnetic poles are placed on the faces. The field falls off rapidly with distance in the x-direction, and smoothly changes direction, except near the corner of the magnet C, until on the symmetry axis OX it is in the opposite direction to **M**.

There are several exercises that can be performed with the computer program given here, simply by changing the input parameters. For instance it is instructive to investigate how the magnetic field distribution depends on the various constants, and how the convergence of the iterative process depends on α. Four typical exercises of this type are:

(1) Determination of the dependence of the magnetic field distribution on the size, shape, and magnetization of the magnet (vary input parameters IN, IM, JN, JM and AMAG).

(2) Investigation of the effects of changing the mesh size h and the determination of the optimum size for h (vary H in conjunction with IN, IM, JN and JM to retain the same region and magnet sizes).

(3) Investigation of the sensitivity of the magnetic field distribution near the magnet to the distance of the outer boundary from the magnet (vary IN and IM). Ideally the far boundary should be at infinity.

(4) Determination of the dependence of the rate of convergence of the iterative process for the $\phi_{i,j}$ on the value of the successive over-relaxation parameter (vary ALPHA) (cf. ref. 3 for the theory).

More insight into the method of solution of the problem can be obtained by making modifications to the program. Five typical exercises, involving only minor changes to the coding, could be based on using the following:

(a) other convergence parameters EPMAX and EPSUM;

(b) changing the coding to calculate the magnetic induction field **B** (from equation (5));

(c) more complicated difference formulae;[4]

(d) better formulae for special points,[4] e.g. for point C;

(e) other convergence criteria.[3,4]

The next stage of investigation would involve considering different models requiring considerable changes to the basic formulae, and effectively requiring completely new programs to be written. Exercises of this type are more advanced and could form the basis of project work. Typical examples include the consideration of the following:

(i) magnets of different shapes (but with plane boundaries);

(ii) composite rectangular magnets of materials with different magnetizations;

(iii) the effect of using a different mesh size over some part of the region of the field;

(iv) two or more separated rectangular magnets;

(v) cylindrical magnets.

Having moved away from finite difference formulae in Cartesian coordinates, as in (v) above, many other projects are possible. There are also problems in many other areas of physics that require solutions to elliptical partial differential equations obtainable by a similar numerical treatment to that considered here.

REFERENCES

1. W. K. H. Panofsky and M. Philips, *Classical Electricity and Magnetism* (Addison-Wesley, Reading, Mass., 1962).
2. C. D. Smith, *Numerical Solution of Partial Differential Equations* (Oxford University Press, London, 1965).
3. *Modern Computing Methods*, 2nd edn. (H.M.S.O., London, 1961).
4. K. J. Binns and P. J. Lawrenson, *Analysis and Computation of Electric and Magnetic Field Problems* (Pergamon Press, London, 1963).

PERMANENT MAGNETS

```
      DIMENSION PHI(50,50),TH(50,50),FE(50,50)
C     INPUT AND PRINT OUT PARAMETERS
      READ(1,100)IN,JN,IM,JM,H,ALPHA
      READ(1,101)AMAG
  100 FORMAT(4I4,2F5.2)
  101 FORMAT(F5.2)
      WRITE(1,105)IN,JN
      WRITE(1,106)IM,JM
      WRITE(1,107)H
      WRITE(1,108)AMAG
      WRITE(1,109)ALPHA
  105 FORMAT('REGION SIZE=',I2,'X',I2)
  106 FORMAT('MAGNET SIZE=',I2,'X',I2)
  107 FORMAT('MESH INTERVAL=',F5.2)
  108 FORMAT('MAGNETIZATION=',F5.2)
  109 FORMAT('CONVERGENCE FACTOR=',F5.2)
      WRITE(1,221)
      WRITE(1,221)
      WRITE(1,222)
  222 FORMAT(5X,'MESH USED')
      WRITE(1,221)
      EPMAX=1.0E-04
      EPSUM=1.0E-04
      HAM=H*AMAG
C     PRINT MESH PATTERN
      CALL MESH(IN,JN,IM,JM)
      INL=IN-1
      JNL=JN-1
C     SET OUTER BOUNDARIES
      DO 1 J=1,JN
      PHI(1,J)=0.0
    1 PHI(IN,J)=0.0
      DO 2 I=1,IN
      PHI(I,1)=0.0
    2 PHI(I,JN)=0.0
C     SET TRIAL INITIAL VALUES
      DO 5 I=2,INL
      DO 6 J=2,JNL
    6 PHI(I,J)=0.5
    5 CONTINUE
      ALPF=ALPHA/4.0
C     BEGIN ITERATION LOOP
      NCOUN=0
   10 NCOUN=NCOUN+1
      IF(NCOUN.GT.100) GOTO 11
      REMAX=0.0
      RESUM=0.0
      DO 3 J=1,JNL
      DO 4 I=1,INL
      IF((I.EQ.1).AND.(J.EQ.1)) GOTO 26
      IF((I.EQ.1).AND.(J.EQ.JM)) GOTO 23
      IF(I.EQ.1) GOTO 20
      IF(J.EQ.1) GOTO 21
      IF((I.EQ.IM).AND.(J.LT.JM)) GOTO 22
      IF((J.EQ.JM).AND.(I.LT.IM)) GOTO 23
      IF((J.EQ.JM).AND.(I.EQ.IM)) GOTO 24
```

PERMANENT MAGNETS

```
      R=PHI(I+1,J)+PHI(I-1,J)+PHI(I,J-1)+PHI(I,J+1)-4.0*PHI(I,J)
      GOTO 25
   20 R=2.0*PHI(2,J)+PHI(1,J-1)+PHI(1,J+1)-4.0*PHI(1,J)
      GOTO 25
   21 R=PHI(I+1,1)+PHI(I-1,1)-2.0*PHI(I,1)
      GOTO 25
   22 R=2.0*(PHI(IM+1,J)+PHI(IM-1,J))-4.0*PHI(IM,J)
      GOTO 25
   23 R=2.0*(PHI(I,JM+1)+PHI(I,JM-1)+HAM)-4.0*PHI(I,JM)
      GOTO 25
   24 R=PHI(IM+1,JM)+PHI(IM-1,JM)+PHI(IM,JM+1)+PHI(IM,JM-1)+
     1HAM-4.0*PHI(IM,JM)
      GOTO 25
   26 R=0.0
   25 CONTINUE
      PHI(I,J)=PHI(I,J)+ALPF*R
      AR=ABS(R)
      IF(AR.GT.REMAX) REMAX=AR
      RESUM=RESUM+R*R
    4 CONTINUE
    3 CONTINUE
C    TEST CONVERGENCE
      RESUM=SQRT(RESUM/FLOAT(INL*JNL))
      IF(REMAX.LT.EPMAX) GOTO 12
      IF(RESUM.GT.EPSUM) GOTO 10
   12 CONTINUE
   11 CONTINUE
      WRITE(1,120)NCOUN,REMAX,RESUM
  120 FORMAT(1H ,' NO OF ITERATIONS =',I3,'   REMAX= ',
     1E10.4,'       RESUM=',E10.4)
      WRITE(1,221)
      WRITE(1,221)
      WRITE(1,121)
  121 FORMAT(1H ,15X,'MAGNETOSTATIC POTENTIAL',)
      WRITE(1,221)
  221 FORMAT(1H )
C    CALCULATE MAGNETIC FIELD AND OUTPUT ARRAYS
      CALL OUT(IN,JN,PHI,9,3)
      CALL THEMA(IN,JN,H,PHI,TH,FE)
      WRITE(1,122)
  122 FORMAT(15X,'MAGNETIC FIELD DIRECTIONS',)
      WRITE(1,221)
      CALL OUT(IN,JN,TH,9,2)
      WRITE(1,123)
  123 FORMAT(15X,'MAGNETIC FIELD MAGNITUDES',)
      WRITE(1,221)
      CALL OUT(IN,JN,FE,9,2)
  111 FORMAT(1H ,5E10.2)
      STOP
      END
      SUBROUTINE OUT(IN,JN,F,NPRIN,NCON)
      DIMENSION F(50,50)
C    REORDERS ARRAY F AND PRINTS NPRIN COLUMNS AT A TIME
C    IF NCON=1 OUTPUTS CHARACTERS FOR MESH
      IP=0
    1 IP=IP+NPRIN
```

```
      INP=IN
      IF(IP.LT.IN) INP=IP
      IL=IP-NPRIN+1
      JLN=1
      DO 2 JT=JLN,JN
      J=JN-JT+1
      IF(NCON.EQ.1) GOTO 3
      WRITE(1,111)(F(IK,J),IK=IL,INP)
      GOTO 2
    3 WRITE(1,113)(F(IK,J),IK=IL,INP)
    2 CONTINUE
      WRITE(1,112)
      WRITE(1,112)
      WRITE(1,112)
      IF(IP.LT.IN) GOTO 1
  111 FORMAT(1H ,12F7.2)
  112 FORMAT(1H )
  113 FORMAT(1H ,20A4)
      RETURN
      END
      SUBROUTINE MESH(IN,JN,IM,JM)
      DIMENSION Q(50,50)
      INTEGER T1,T2
C     SETS ARRAY OF CHARACTERS FOR MESH REGIONS
      DATA T1,T2/1H.,1H1/
      DO 1 I=1,IN
      DO 2 J=1,JN
      Q(I,J)=T1
    2 IF((I.LE.IM).AND.(J.LE.JM)) Q(I,J)=T2
    1 CONTINUE
      CALL OUT(IN,JN,Q,20,1)
      RETURN
      END
      SUBROUTINE THEMA(IN,JN,H,F,T,HM)
      DIMENSION F(50,50),HM(50,50),T(50,50)
C     INPUTS POTENTIAL AS F AND OUTPUTS MAGNETIC FIELD IN
C     ARRAYS T(DIRECTIONS) AND HM(MAGNITUDES)
      DO 1 I=2,IN
      DO 2 J=2,JN
      HX=-(F(I,J)-F(I-1,J))/H
      HY=-(F(I,J)-F(I,J-1))/H
      IF(HX.EQ.0.0) HX=0.01
      HXY=HY/HX
      TT=ATAN(ABS(HXY))*180.0/3.14159
C     ADJUST ARCTAN FOR DIFFERENT QUADRANTS
      IF((HX.GE.0.0).AND.(HY.GE.0.0)) P=TT
      IF((HX.LT.0.0).AND.(HY.GE.0.0)) P=180.0-TT
      IF((HX.LE.0.0).AND.(HY.LT.0.0)) P=180.0+TT
      IF((HX.GE.0.0).AND.(HY.LT.0.0)) P=360.0-TT
      T(I,J)=P
      HM(I,J)=SQRT(HX*HX+HY*HY)
    2 CONTINUE
    1 CONTINUE
      DO 3 J=2,JN
      HY=-(F(1,J)-F(1,J-1))/H
      T(1,J)=90.0
```