

Figure 14. Hologram of the letter-group object MESC: without random phasing

For the object MESC holograms, such as those shown in Figures 14 and 15, are produced according to whether or not random phasing is included. These holograms cover an 80×80 unit cell array using AMULT$=0.66$ and the unmodified clipping factor, set inside the program. Both holograms exhibit the star-like structure associated with the lines of the object. Figure 15 shows quite clearly the way in which the random phasing 'spreads' the picture out.

7. THE RECONSTRUCTION

The computer-generated holograms shown in Figures 14 and 15 are Fourier transform holograms.[1,7,11] The reconstruction process is, in principle, the



Figure 15. Hologram of the letter-group object MESC: with random phasing

optical process shown in Figure 16. An optically produced Fourier transform hologram uses the Fourier transforms of the object and reference source to produce an interference pattern. The reconstruction process produces the usual twin images, but in the special case of the Fourier transform hologram they are located at infinity and a lens is normally used to relocate them in its focal plane. The twin images are related to each other by an inversion operation, through the zero order position.

The computer-generated hologram behaves in just the same way, with respect to the twin images and the inversion, but sampling the true Fourier transform leads to spectra, i.e. many other images are present. Optically produced holograms do not have this feature.



Figure 16. Reconstruction with a Fourier transform hologram

The results, obtained by reconstructing the random phase hologram, generated by the computer program given here, are shown in the photographs in Figures 17 and 18. These are obtained from holograms consisting of photographically reduced graph plotter output. Very good results are also obtained by using the 35 mm slide film directly produced by the computer, without any further photographic work. A He–Ne laser was used by focusing it to a point source with an ordinary microscope lens of 25 mm focal length. It was then collimated to a plane wave of sufficient cross-sectional area to illuminate the hologram. The reconstructions are generally small so it was found easier to observe the images on a screen in the far field, at several metres, rather than use a second lens. A positive or a negative of the hologram works equally well.

Figure 17 shows clearly the original MESC object and also that sampling has the effect of producing other images, even though the light intensity in them falls away quite rapidly. The effect of oversampling is shown by the fact that the images are nicely separated. Figure 18 is a novel photograph obtained by allowing the laser beam to enter the camera directly. Thus Figure 18 is an aerial photograph in which the objective lens is used to mask off the rest of the image field.

8. EXERCISES AND PROJECT SUGGESTIONS

The program can be used as a demonstration to produce holograms and reconstructions of the initials or name of the user. A deeper investigation will lead to an understanding of the effects of oversampling and undersampling, but it should be noted that certain straight-line objects, such as the



Figure 17. Photograph of the image field obtained using the hologram in Figure 15 produced directly on microfilm

letter H, for instance, have only vertical and horizontal lines. Without random phasing the hologram, for such a letter, has the appearance of a vertical line at right angles to a horizontal line. The introduction of random phasing will split and separate the vertical line into two lines, but the horizontal line is only shifted. A way out of this is to express each line as the sum of N lines where N is large enough to make a difference. If this is done then random phasing will produce many lines in the hologram. This procedure is obviously valuable for general objects and rounded letters that are the limit of an N-line object as N becomes large.

Other investigations could include varying the clipping, taking another look at the approximations, windowing the hologram and considering other apertures and shapes such as circular or elliptic. Finally a line printer was used recently[12] to produce binary holograms. This was done by using the real part of the Fourier transform to control the overprinting, rather than just the amplitude, as is used in the program here. Overprinting is allowed if the real part of the transform is greater than zero, if it is not, then a blank space is left. The computer program given here could be adapted for this purpose by altering the overprinting section and selecting device $1 = LP$ in the data.



Figure 18. Aerial photograph of reconstructed object

REFERENCES

1. R. J. Collier, C. B. Burckhardt, and L. H. Lin, *Optical Holography* (Academic Press, New York, 1971).
2. D. C. Chu and J. R. Fienup, *Opt. Eng.*, **13,** 189 (1974).
3. J. N. Butters, *Holography and its Technology* (Peter Peregrinus Ltd. on behalf of I.E.E., 1971).
4. M. Born and E. Wolf, *Principles of Optics* (Pergamon Press, Oxford, New York, 1970).
5. J. S. Marsh and R. C. Smith, *Am. J. Phys.*, **44,** 774 (1974).
6. R. C. Smith and J. S. Marsh, *J. Opt. Soc.*, **64,** 798 (1974).
7. J. W. Goodman, *Introduction to Fourier Optics* (McGraw-Hill, New York, 1968).
8. W. H. Lee, *App. Opt.*, **9,** 639 (1970).
9. B. R. Brown and A. W. Lohmann, *IBM J. Res. Dev.*, **13,** 160 (1969).
10. A. W. Lohmann and D. P. Paris, *App. Opt.*, **6,** 1739 (1967).
11. T. S. Huang, *I.E.E.E.*, **59,** 1335 (1971).
12. K. Nagashima, *Opt. Comm.*, **17,** 273 (1976).

DIGITAL HOLOGRAMS

```
C     THIS PROGRAM GENERATES DIGITAL HOLOGRAMS FOR OBJECTS
C     MADE UP OF STRAIGHT LINES. IT IS WRITTEN
C     TO RUN INTERACTIVELY(OR  OTHERWISE) IN STANDARD FORTRAN. THE
C     GRAPHICAL PROCEDURES USE ELEMENTS OF GINO-F(GRAPHICAL INPUT/OUTPUT
C     -FORTRAN)DEVELOPED BY THE COMPUTER AIDED DESIGN CENTRE ,
C     CAMBRIDGE UNIVERSITY.IT IS READILY AVAILABLE ON DIFFERENT
C     MACHINES IN THE U.K. AND ELSEWHERE.UPTO 30 LINES MAYBE
C     SPECIFIED BY THEIR CO-ORDINATES IN AN AREA DEFINED BY
C      -10.0<X<10.0 , -7.5<Y<7.5 IN ARBITRARY UNITS.THE HOLOGRAM CAN
C     BE AS LARGE AS 80 BY 80 CELLS-THIS IS SET TO CONTAIN
C     PROGRAM WITHIN A CORE REQUIREMENT OF 80K

C     DEFINITIONS OF CHANNELS USED IN
C     FORMAT STATEMENTS.
C        CHANNEL 1=CONTROLFILE-INPUT
C        CHANNEL 2=OUTPUTS REQUESTS FOR DATA IN AN INTERACTIVE
C        MODE-FILED IF NOT REQUIRED
C        CHANNEL 3=RESULTS-OUTPUTS AMPLITUDE AND PHASE DISTRIBUTION.
C        CHANNEL 4=OBJECT-INPUTS

C     DEFINITION OF VARIABLES
C       FXY(X,Y)       = AMPLITUDE OF FOURIER TRANSFORM
C       ANGLE(X,Y)     = PHASE OF FOURIER TRANSFORM
C       F              = COMPLEX FOURIER TRANSFORM
C       APOINT         = ARRAY CONTAINING CO-ORDINATES
C                        OF LINE ENDS
C       COSTH,SINTH    = COSINE AND SINE OF ANGLE LINE
C                        MAKES WITH X-AXIS
C       L              = ARRAY CONTAINING LINE LENGTHS
C       A,B            = ARRAYS USED FOR RANDOM PHASE
C       XM,YM          = ARRAYS CONTAINING CO-ORDINATES
C                        OF MID-POINTS OF LINE SEGMENTS
C     FREE FORMATS ARE NOT USED BECAUSE THEY ARE NOT AVAILABLE
C     IN THE DESIRED FORM ON ALL MACHINES

      REAL L
      DIMENSION FXY(80,80),ANGLE(80,80),APOINT(30,4),
     1COSTH(30),SINTH(30),L(30),A(30),B(30)
      DIMENSION XM(30),YM(30)
      COMMON FXY,ANGLE,APOINT
      COMPLEX F
C     SET ARRAYS AND CONSTANTS
      DO 9 I=1,30
      A(I)=0.0
      B(I)=0.0
    9 CONTINUE
C     STARTING VALUES
      XMAX=-100.0
      XMIN=100.0
      YMAX=-100.0
      YMIN=100.0
      PI=4.0*ATAN(1.0)
      WRITE(2,300)
  300 FORMAT(1X,'TYPE OUTPUT DEVICE 1=LP,2=NARROW,3=WIDE,4=FILM'/)
      READ(1,301)NDEV
  301 FORMAT(I1)
```

DIGITAL HOLOGRAMS

```
C     THE NUMBER OF LINES THEN THEIR COORDINATES ARE READ IN

      WRITE(2,99)
  99  FORMAT(1X,'TYPE NUMBER OF LINE SEGMENTS --MAX 30 I2 FORM'/)
      READ(4,21)NULINE
  21  FORMAT(I2)
      WRITE(2,100)
 100  FORMAT(1X,'SPECIFY ALL LINE SEGMENTS FROM LEFT TO RIGHT'/)
      DO 3 I=1,NULINE
      WRITE(2,101)
 101  FORMAT(1X,'TYPE -- X1,Y1,X2,Y2 IN 4(F4.1,1X) FORM'/)
      READ(4,200)(APOINT(I,II),II=1,4)
 200  FORMAT(4(F4.1,1X))

C     MAXIMUM AND MINIMUM CO-ORDINATES
C     IN X AND Y ARE CALCULATED
C     CALCULATE THE LINE LENGTHS AND ALSO THE COSINE AND SINE
C     OF THE ANGLE THE LINE MAKES WITH THE X- AXIS

      IF(APOINT(I,3).GT.XMAX)XMAX=APOINT(I,3)
      IF(APOINT(I,1).LT.XMIN)XMIN=APOINT(I,1)
      IF(APOINT(I,2).GE.APOINT(I,4))GOTO 5
      IF(APOINT(I,4).GT.YMAX)YMAX=APOINT(I,4)
      IF(APOINT(I,2).LT.YMIN)YMIN=APOINT(I,2)
      GOTO 6
   5  IF(APOINT(I,2).GT.YMAX)YMAX=APOINT(I,2)
      IF(APOINT(I,4).LT.YMIN)YMIN=APOINT(I,4)
   6  L(I)=SQRT((APOINT(I,3)-APOINT(I,1))**2+(APOINT(I,4)-APOINT
     1(I,2))**2)
      COSTH(I)=ABS(APOINT(I,3)-APOINT(I,1))/L(I)
      SINTH(I)=ABS(APOINT(I,4)-APOINT(I,2))/L(I)
      IF(APOINT(I,4).LT.APOINT(I,2))COSTH(I)=-COSTH(I)
      XM(I)=ABS((APOINT(I,1)-APOINT(I,3))/2.0)+APOINT(I,1)
      YM(I)=ABS((APOINT(I,2)-APOINT(I,4))/2.0)
      IF(APOINT(I,2).LE.APOINT(I,4))YM(I)=YM(I)+APOINT(I,2)
      IF(APOINT(I,4).LT.APOINT(I,2))YM(I)=YM(I)+APOINT(I,4)
   3  CONTINUE

C     HERE THE PROGRAM CALCULATES THE BANDWIDTH
C     IN BOTH DIRECTIONS(X AND Y).IT THEN SETS THE CRITICAL
C     SAMPLING RATE AS 2.0*PI/(LARGEST VALUE)

      IF(XMAX-XMIN.GE.YMAX-YMIN)SAMPLE=2.0*PI/(XMAX-XMIN)
      IF(YMAX-YMIN.GT.XMAX-XMIN)SAMPLE=2.0*PI/(YMAX-YMIN)
      WRITE(3,202)SAMPLE
 202  FORMAT(1X,'SAMPLE RATE = ',F7.3)

C     OVERSAMPLING LEADS TO
C     A  RECONSTRUCTION WITH IMAGES NICELY SEPARATED
C     BUT HERE THE SAMPLING RATE CAN BE MODIFIED
C     TO ALLOW BOTH UNDER AND OVER
C     SAMPLING TO BE INVESTIGATED

      WRITE(2,203)
```

DIGITAL HOLOGRAMS

```
 203  FORMAT(1X,'DO YOU WISH TO MODIFY THE SAMPLING RATE,'/
     1' TYPE MULTIPLICATION FACTOR IN F4.2 FORM'/)
      READ(1,204)AMULT
 204  FORMAT(F4.2)
      SAMPLE=SAMPLE*AMULT
      WRITE(3,205)SAMPLE
 205  FORMAT(1X,'MODIFIED SAMPLE RATE = ',F7.3)
      WRITE(2,103)
 103  FORMAT(1X,'HOW MANY SAMPLES IN X — MAX 80 I3 FORM'/)
      READ(1,201)ISAMPS
      WRITE(2,104)
 104  FORMAT(1X,'HOW MANY SAMPLES IN Y — MAX 80 I3 FORM'/)
      READ(1,201)JSAMPS
 201  FORMAT(I3)

C     THE PROGRAM CALCULATES THE ANALYTICAL FOURIER TRANSFORM
C     FOR EACH CELL

      WRITE(2,105)
 105  FORMAT(1X,'TO EQUALISE AMPLITUDE DISTRIBUTION',
     1' TYPE 1 ELSE 0')
      READ(1,106)NRAN
 106  FORMAT(I1)

C     TO EVEN OUT AMPLITUDES IN HOLOGRAM THE PHASE OF EACH
C     ELEMENT IN THE OBJECT IS ALLOWED TO VARY LINEARLY
C     ALONG ITS LENGTH — IF NRAN IS SET TO 1

      IF(NRAN.EQ.0)GOTO 10
      X=(FLOAT(ISAMPS)/2.0-0.5)*SAMPLE
      Y=(FLOAT(JSAMPS)/2.0-0.5)*SAMPLE

      DO 8 I=1,NULINE

C     G05AAF(X) IS A ROUTINE THAT PRODUCES
C     RANDOM NUMBERS BETWEEN 0 AND 1

      A(I)=(G05AAF(XXX)*2.0-1.0)*X/4.0
      B(I)=(G05AAF(YYY)*2.0-1.0)*Y/4.0
    8 CONTINUE

C     SUM THE FOURIER TRANSFORM OF EACH LINE
C     AT EACH SAMPLING POINT

  10  DO 1 J=1,JSAMPS
      Y=(-(FLOAT(JSAMPS)/2.0+0.5)+FLOAT(J))*SAMPLE
      DO 2 I=1,ISAMPS
      X=(-(FLOAT(ISAMPS)/2.0+0.5)+FLOAT(I))*SAMPLE
   7  F=CMPLX(0.0,0.0)
      DO 20 NFOUR=1,NULINE
      SINC=L(NFOUR)/2.0
      SINC=SINC*((X-A(NFOUR))*COSTH(NFOUR)+(Y-B(NFOUR))*SINTH(NFOUR))

C     A CHECK TO STOP OVERFLOW I.E. 0.0/0.0

      IF(ABS(SINC).GT.10.0**(-20))GOTO 50
```

DIGITAL HOLOGRAMS

```
      SINC=1.0
      GOTO 51
   50 SINC=SIN(SINC)/SINC
   51 SINC=L(NFOUR)*SINC
      F=F+CEXP(CMPLX(0.0,(X-A(NFOUR))*XM(NFOUR)+(Y-B(NFOUR))*YM(NFOUR)))
     1*CMPLX(SINC,0.0)
   20 CONTINUE

C     THE FOURIER TRANSFORM IS OF COMPLEX FORM AND IS NOW SPLIT
C     INTO AN AMPLITUDE TERM AND A PHASE TERM. THE PHASE TERM
C     IS NORMALISED OVER THE INTERVAL 0 TO 2*PI.

      FXY(I,J)=CABS(F)
      BB=AIMAG(F)
      AA=REAL(F)
      ANGLE(I,J)=ATAN2(BB,AA)/(2.0*PI)
    2 CONTINUE
    1 CONTINUE
      WRITE(2,108)
  108 FORMAT(1X,'AMPLITUDE AND NORMALISED PHASE CALCULATED')
      CALL AOBJCT(NULINE,NDEV)
      WRITE(2,109)
  109 FORMAT(1X,'INPUT PLOTTED')
      CALL RELACS(ISAMPS,JSAMPS,NDEV,CLIP)
      CALL DISPLAY(ISAMPS,JSAMPS,NDEV,CLIP)

C     GINO-F SUBROUTINE - DEVFIN TERMINATES
C     THE GRAPHIC DEVICE USED-SWITCHES OFF GLOBAL CALLS

      CALL DEVFIN
      STOP
      END


      SUBROUTINE AOBJCT(NULINE,NDEV)

C     THIS SUBROUTINE PLOTS THE INPUTTED OBJECT
C     AND PROVIDES A CHECK OF THE DATA.IT COULD BE OMITTED.

      DIMENSION APOINT(30,4),FXY(80,80),ANGLE(80,80)
      COMMON FXY,ANGLE,APOINT

C     ANGLE IS REDUNDANT HERE.IT IS KEPT FOR CONVENIENCE TO
C     PRESERVE THE FORM OF THE COMMON STATEMENT.
C     GINO-F SUBROUTINES
C        NARROW   - CALLS NARROW PAPER CALCOMP
C        PRINTR   - CALLS LINE PRINTER
C        WIDE     - CALLS WIDE PAPER CALCOMP
C        FILM     - CALLS 35MM SLIDE PLOT-PRODUCES HOLOGRAMS DIRECTLY.
C        PENSEL   - SELECTS LINE THICKNESS ON FILM
C        UNITS(S) - S=NUMBER OF MM'S IN CURRENT DRAWING UNITS
C        WINDO2   - SETS UP 2-D WINDOW
C        SHIFT2   - SHIFTS REFERENCE AXIS BY VECTOR INCREMENT (X,Y)
C        MOVTO2   - POSITION PEN-BEAM AT A POINT X,Y
C        LINTO2   - DRAW A 2-D LINE FROM CURRENT POSITION TO X,Y
C        MOVBY2   - POSITION THE PEN -BEAM (2-D) (A,B)
```

DIGITAL HOLOGRAMS

```
C                     INCREMENTAL CO-ORDINATE DISTANCE
C         LINBY2    - DRAW A 2-D LINE FROM THE CURRENT POSITION
C                     POSITION (A,B) INCREMENTAL CO-ORDINATE DISTANCE
C         PICCLE    - CLEARS DRAWING AREA OF ALL PREVIOUS PICTURES
C         TRANSF(0)- SWITCHES OFF SHIFT TRANSFORMATION
C      PAPENQ(X,Y,C)-DRAWING AREA.X,Y:- PAPER SIZE.C:- PAPER TYPE.
C      DEVPAP(A,B,C)-SPEC. OF  PLOTTER PAPER,DIMENSIONS A*B,TYPE:- C.
C      C IS DUMMY IF NO CHOICE PAPER AVAILABLE

       IF(NDEV.EQ.1)CALL PRINTR
       IF(NDEV.EQ.2)CALL NARROW
       IF(NDEV.EQ.3)CALL WIDE
       IF(NDEV.EQ.4) CALL FILM
C      NOTE:CALLS TO OUTPUT DEVICES ARE GLOBAL.
       CALL UNITS(10.0)
C      DRAWING UNITS ARE NOW CM
       IF(NDEV.EQ.4)CALL PENSEL(1,0.15,1)
C      COLOUR,LINEWIDTH,TYPE-FIRST AND LAST ARE IRREVELANT FOR FILM
       CALL PAPENQ(XPAP,YPAP,IP)
       CALL WINDO2(0.0,20.0,0.0,15.0)
C      WINDOW DEFINED XLEFT,XRIGHT,YBOTTOM,YTOP
       CALL SHIFT2(10.0,7.5)
       CALL MOVTO2(0.0,7.5)
       CALL LINBY2(0.0,-15.0)
       CALL MOVTO2(-10.0,0.0)
       CALL LINBY2(20.0,0.0)
       DO 3 I=1,NULINE
       CALL MOVTO2(APOINT(I,1),APOINT(I,2))
       CALL LINTO2(APOINT(I,3),APOINT(I,4))
     3 CONTINUE
       CALL WINDO2(0.,XPAP,0.,YPAP)
C      RESET WINDOW TO FULL PAPER SIZE
       CALL TRANSF(0)
       CALL PICCLE
       RETURN
       END

       SUBROUTINE RELACS(NXDIM,NYDIM,NDEV,CLIP)

C      THIS SUBROUTINE NORMALISES THE AMPLITUDE TO 1
C      IT ALSO GENERATES A HISTOGRAM OF BOTH THE NORMALISED
C      AMPLITUDE AND PHASE TO AID CHOOSING A CLIPPING
C      FACTOR IN THE DISPLAYING OF THE HOLOGRAM.THE AMPLITUDE
C      DISTRIBUTION IS  CLIPPED TO 99 PER CENT,
C      UNLESS OTHERWISE ALTERED.

C      VARIABLE DEFINITION
C        HISTA    - ARRAY CONTAINING NUMBER AT COUNTA
C        COUNTA   - ARRAY CONTAINING AMPLITUDE LEVELS
C        HISTP    - ARRAY CONTAINING NUMBER AT COUNTP
C        COUNTP   - ARRAY CONTAINING PHASE LEVELS
C        ITA      - ARRAY CONTAINING TITLES FOR AMPLITUDE HISTOGRAM
C        ITP      - ARRAY CONTAINING TITLES FOR PHASE HISTOGRAM

      DIMENSION FXY(80,80),HISTA(50),COUNTA(50),ITA(15),ANGLE(80,80)
      DIMENSION APOINT(30,4),HISTP(50),COUNTP(50),ITP(14)
```

DIGITAL HOLOGRAMS

```
C     SCDTA IS REQUIRED BY GRAPH PLOTTING SUBROUTINE FGPLT

      COMMON/SCDTA/RV(1)
      COMMON FXY,ANGLE,APOINT
      DATA ITA/34,'AMPLITUDE DISTRIBUTION IN HOLOGRAM',3,'AMP',
     15,'COUNT'/
      DATA ITP/30,'PHASE DISTRIBUTION IN HOLOGRAM',3,'PHA',5,'COUNT'/

C     RESET UNITS

      CALL UNITS(1.)
      LEVELS=50
C     NUMBER OF HISTOGRAM CHANNELS

C     SETS UP INITIAL VALUES OF HISTA,HISTP,COUNTA,COUNTP


C     SEARCH FXY FOR MAXIMUM VALUE THEN NORMALISE FXY TO 1.
C     CONSTRUCT AMPLITUDE AND PHASE HISTOGRAMS

      DO 10 K=1,LEVELS
      HISTA(K)=0.0
      HISTP(K)=0.0
      COUNTA(K)=FLOAT(K)/FLOAT(LEVELS)
      COUNTP(K)=COUNTA(K)-0.5
   10 CONTINUE
      AMAX=0.0
      DO 1 J=1,NYDIM
      DO 1 I=1,NXDIM
      IF(FXY(I,J).GT.AMAX)AMAX=FXY(I,J)
    1 CONTINUE
      DO 2 J=1,NYDIM
      DO 2 I=1,NXDIM
      FXY(I,J)=FXY(I,J)/AMAX
      DO 3 K=1,LEVELS
      IF(FXY(I,J).GT.COUNTA(K))GOTO 3
      HISTA(K)=HISTA(K)+1.0
      GOTO 6
    3 CONTINUE
    6 DO 5 K=1,LEVELS
      IF(ANGLE(I,J).GT.COUNTP(K))GOTO 5
      HISTP(K)=HISTP(K)+1
      GOTO 2
    5 CONTINUE
    2 CONTINUE
      WRITE(3,102)
      WRITE(3,103)


C     MAKE HISTOGRAM MARKERS CENTRAL FOR USE WITH FGPLT

      DO 8 K=1,LEVELS
      X=1.0/FLOAT(LEVELS)
      COUNTA(K)=COUNTA(K)-X/2.0
      COUNTP(K)=COUNTP(K)-X/2.0
    8 CONTINUE
```

```
DIGITAL HOLOGRAMS

C     WRITE OUT VALUES IN TABLE FORM

      DO 4 K=1,LEVELS
      WRITE(3,101)COUNTA(K),HISTA(K),COUNTP(K),HISTP(K)
 101  FORMAT(1X,F5.3,4X,F6.1,5X,F5.2,7X,F5.1)
   4  CONTINUE
 102  FORMAT(1X,'DISTRIBUTION OF AMPLITUDE AND PHASE IN HOLOGRAM'/)
 103  FORMAT(2X,'AMP',4X,'NUMBER',5X,'PHASE',9X,'NUMBER')
      WRITE(2,100)
 100  FORMAT(1X,'AMPLITUDE NORMALISED'/)
      DENOM=FLOAT(NXDIM*NYDIM)
      TOTAL=0.0
C     DEFINE A CLIPPING FACTOR TO EFFECTIVELY

C     CLIP ONE PER CENT OF POINTS OFF TOP END OF AMPLITUDE DISTRIBUTION
C          IF THIS IS NOT THE BEST CLIP THEN THE USER IS ALLOWED TO
C     ADJUST IT.

      DO 9 I=1,LEVELS
      TOTAL=HISTA(I)+TOTAL
      IF(TOTAL/DENOM.LT.0.99)GOTO 9
      CLIP=FLOAT(LEVELS)/FLOAT(I)
      WRITE(3,105)CLIP
      GOTO 11
   9  CONTINUE
 105  FORMAT(1X,'SUGGESTED CLIP = ',F6.3, 'TO CHANGE TYPE 1 ELSE 0')
   7  FORMAT(I1)
  11  READ(1,106)NCLIP
 106  FORMAT(I1)
      IF(NCLIP.EQ.0)GOTO 12
      WRITE(3,107)
 107  FORMAT(1X,'TYPE NEW CLIP VALUE F4.2 FORM'/)
      READ(1,108)CLIP
 108  FORMAT(F4.2)

C     GINO-F SUBROUTINES
C        A4       - SETS PLOT TO A4 SIZE (29 CM*21 CM)
C        FGPLT    - PLOTS A HISTOGRAM WHERE
C                       HISTA IS THE NUMBER OF POINTS
C                       AT COUNTA - THESE ARE BOTH REAL ARRAYS
C                       OF SIZE LEVELS
C        NEWPAG   - CALLS A NEW PAGE

  12  CALL A4

C     HISTOGRAMS OF AMPLITUDE AND PHASE DISTRIBUTIONS
C     FGPLT IS A STANDARD PLOTTING ROUTINE

      CALL FGPLT(COUNTA,HISTA,LEVELS,6,0,1,0,ITA)
      CALL FGPLT(COUNTP,HISTP,LEVELS,6,0,1,0,ITP)
      CALL NEWPAG
      RETURN
      END
```

DIGITAL HOLOGRAMS

```
      SUBROUTINE DISPLAY(NXDIM,NYDIM,NDEV,CLIP)
C     THIS SUBROUTINE DRIVES THE O/P  DEVICE
C     DRAWING THE HOLOGRAM
C     WHEN USING THE FILM OUTPUT IT IS DIRECTLY
C     USABLE AS A HOLOGRAM IN AN OPTICAL RECONSTRUCTION.
C     BOTH THE NARROW AND WIDE CALCOMP PLOTTER OUTPUT
C      WILL REQUIRE PHOTOGRAPHIC REDUCTION-THE PHOTOGRAPHIC
C     FILM THEN BEING USED AS THE HOLOGRAM.
C     THE LINEPRINTER OUTPUT IS JUST AN AMPLITUDE
C     PLOT AND IS INCLUDED TO PROVIDE IMMEDIATE OUTPUT
C     SO THAT THE USER CAN CHECK THE CLIPPING CHOSEN.

C     VARIABLE DEFINITION
C        CHAR   - INTEGER ARRAY CONTAINING CODE
C                 FOR SYMBOL OUTPUT ON QUICK LINEPRINTER
C                 OVERPRINTING AMPLITUDE PLOT

      INTEGER CHAR,E
      DIMENSION FXY(80,80),ANGLE(80,80),CHAR(10),E(130),APOINT(30,4)
      COMMON FXY,ANGLE,APOINT
      DATA CHAR/' ',':','-','I','/','+','L','=','U','0'/
      IXX=0
   25 IF(NDEV.EQ.1)GOTO 6
   51 FORMAT(I1)
      WRITE(2,995)
  995 FORMAT(1X,'HOW MANY SMALLER HOLOGRAMS - TYPE NUMBER')
      READ(1,51)NSM

C     NEXT SECTION CALLS O/P DEVICE AND CHOOSES SCALING

C      NUMERATOR OF XS AND YS ARE THE MAX SIZE OF
C      DRAWING AREA IN MM. MINIMUM XS OR YS GIVES
C      MM PER UNIT CELL,IN ORDER TO FILL THE
C      FULL DRAWING AREA.

C      VARIABLE DEFINITION
C         ALINWD - GIVES A VALUE OF THE PEN LINEWIDTH
C         GINO-F SUBROUTINES
C         DEVPAP - DEFINES THE PAPER SIZE

      CALL PICCLE
      IF(NDEV.NE.2)GOTO 20
      XS=3600.0/FLOAT(NXDIM)
      YS=270.0/FLOAT(NYDIM)
      ALINWD=0.5
C     BASED UPON THE ACTUAL PEN WIDTH IN MM.
      CALL DEVPAP(3600.,270.,III)
      GOTO 30
   20 IF(NDEV.NE.3)GOTO 21
      XS=3600.0/FLOAT(NXDIM)
      YS=840.0/FLOAT(NYDIM)
      ALINWD=0.5
      CALL DEVPAP(3600.,840.,III)
```

```
DIGITAL HOLOGRAMS

      GOTO 30
  21  XS=430.0/FLOAT(NXDIM)
      YS=270.0/FLOAT(NYDIM)
      ALINWD=0.2
C     THINNER LINES  ALLOWED ON FILM
      CALL DEVPAP(430.,270.,III)

C     AMIN1 IS A STANDARD FORTRAN FUNCTION TO FIND
C     A MINIMUM VALUE

C     VARIABLE DEFINITION
C        NULINE IS THE NUMBER OF PEN LINES THAT WILL FILL
C        A SAMPLING CELL IN THE OUTPUT HOLOGRAM

  30  S=AMIN1(XS,YS)
      CALL UNITS(S)
      SS=1.0/S
      IF(NDEV.EQ.4)CALL PENSEL(1,SS,1)
C     DIRECT HOLOGRAM PRODUCTION AS 35MM FILM SLIDE
      NULINE=2*IFIX(S/ALINWD+0.5)-1
      WRITE(3,200)S,NULINE
 200  FORMAT(1X,'UNITS = ',F7.3,' NUMBER OF LINES = ',I2/)
      CALL WINDO2(0.0,FLOAT(NXDIM),0.0,FLOAT(NYDIM))

C     THE CENTRE OF THE SQUARE SAMPLING CELL IS THE DATUM
C     THE PEN IS POSITIONED IN THE FIRST CELL IN THE FIRST
C     ROW AT (-0.5,-0.5) WITH RESPECT TO DATUM.


C     THE PEN IS STEPPED FROM SAMPLING CELL TO SAMPLING CELL IN ONE
C     ROW AND * IN EACH CELL COMPARE THE POSITION OF THE PEN WITH THE
C     AMPLITUDE FXY AND DECIDE WHETHER OR NOT TO DRAW
C     A LINE. THEN INCREMENT THE PEN BY HALF A LINE WIDTH(DRAWN BY PEN)
C     AND REPEAT FROM *. WHEN ONE ROW IS COMPLETE STEP TO
C     THE NEXT AND REPEAT. THE LINE IS HALF THE LENGTH OF
      A CELL WIDTH AND ITS CENTRE IS MOVED FROM THE CENTRE
      OF THE CELL BY THE VALUE IN ANGLE.

  57  DO 1 I=1,NXDIM
      DO 2 N=1,NULINE
      DO 3 J=1,NYDIM
      CALL MOVBY2(0.0,0.25)
      X=FLOAT(N)/FLOAT(NULINE)
      X1=FLOAT(N-1)/FLOAT(NULINE)
      IF(X1.LT.0.5*(1.0-(FXY(I,J)*CLIP)))GOTO 4
      IF(X.GT.0.5*(1.0+(FXY(I,J)*CLIP)))GOTO 4
      IF(FXY(I,J)*CLIP.LT.ALINWD/S)GOTO 4
      CALL MOVBY2(0.0,ANGLE(I,J))
      CALL LINBY2(0.0,0.5)
      CALL MOVBY2(0.0,-ANGLE(I,J))
      GOTO 5
   4  CALL MOVBY2(0.0,0.5)
   5  CALL MOVBY2(0.0,0.25)
   3  CONTINUE
      CALL MOVBY2(1.0/FLOAT(NULINE),-FLOAT(NYDIM))
   2  CONTINUE
```

DIGITAL HOLOGRAMS

```
    1 CONTINUE
C      IF MORE(SMALLER) HOLOGRAMS ARE REQUIRED,
C      THIS NEXT SECTION RESETS
C      THE WINDO2 AND DRAWS A SMALLER AREA OF THE PREVIOUS HOLOGRAM-
C      IT SUBTRACTS 10 CELLS  FROM EACH AXIS.THIS IS REPEATED UNTIL
C      THE NUMBER OF SMALLER HOLOGRAMS REQUIRED IS PLOTTED
   53 IF(IXX.EQ.NSM)GOTO 56
      CALL PICCLE
      XLEFT=5.0*FLOAT(NSM-IXX)
      XRIGHT=FLOAT(NXDIM)-XLEFT
      YLEFT=XLEFT
      YRIGHT=FLOAT(NYDIM)-YLEFT
      CALL WINDO2(XLEFT,XRIGHT,YLEFT,YRIGHT)
      IXX=IXX+1
      GOTO 57
   56 GOTO 7
C      OVERPRINTING QUICK AMPLITUDE CHECK ON LINEPRINTER
    6 CONTINUE
      DO 8 I=1,NYDIM
      WRITE(3,101)
  101 FORMAT(1H )
      DO 9 L=1,10
      FLAG=0.0
      DO 10 M=1,NXDIM
      IF(FXY(M,I)*CLIP.LT.FLOAT(L)*0.1)GOTO 11
      E(M)=CHAR(L)
      FLAG=1.0
      GOTO 10
   11 E(M)=CHAR(1)
   10 CONTINUE
      IF(FLAG)12,12,9
    9 WRITE(3,102)(E(K),K=1,NXDIM)
   12 CONTINUE
    8 CONTINUE
      WRITE(3,103)
  102 FORMAT(1H+,130A1)
  103 FORMAT(1X,'FINISHED')
C      THE FINAL SECTION ALLOWS THE CLIPPING OR
C      O/P DEVICE TO BE CHANGED WHILST MONITORING THE HOLOGRAM PRODUCTION
C      IT IS INCLUDED TO ALLOW THE USER TO EXPT WITH
C      DATA ON A VIDEO DISPLAY AND THEN OBTAIN THE
C      HARD COPY OF REQUIRED  HOLOGRAM
    7 WRITE(2,105)
  105 FORMAT(1X,'TO VARY CLIPPING TYPE NEW VALUE ELSE <CR>')
      READ(1,152)CLIP
  152 FORMAT(F3.1)
      IF (CLIP.LT.0.1)GOTO 50
      GOTO 25
```

DIGITAL HOLOGRAMS

```
 50  WRITE(2,106)
106  FORMAT(1X,'TO CHANGE O/P DEV TYPE 1=LP,2=NARROW,3=WIDE,',
    1'4=FILM ELSE <CR>'/)
     READ(1,150)NDEV
150  FORMAT(I1)
     IF(NDEV.EQ.0)RETURN
     GOTO 25
     END
```

PART 2

Magnetism

Physics Programs
Edited by A.D. Boardman

CHAPTER 4

Calculation of the Fields Near Permanent Magnets

M. I. DARBY

1. INTRODUCTION

Partial differential equations are important in almost all branches of physics, and often they can only be solved numerically. Owing to the diversity of boundary conditions and other factors that may apply, it is impracticable to produce computer library routines capable of solving more than one specific type of problem. For this reason it is valuable to have some practical experience of the difficulties involved in applying one of the common numerical techniques in a relatively simple situation.

The problem considered here is the calculation of the magnetic field in the vicinity of a uniformly magnetized rectangular permanent magnet. The magnet is assumed to be infinite in one direction, so that the problem reduces to two dimensions. The basic magnetostatic equations are given in sections 2 and 3. There it is shown that the fields are conveniently written in terms of a scalar magnetostatic potential, which satisfies Laplace's equation, and which is completely determined by the boundary conditions. Sometimes it is possible to obtain an analytical solution for the potential, but usually Laplace's equation must be solved numerically. One method of doing so, and that adopted here, is to replace the partial differential equation with a set of (linear) finite difference equations. These can then be solved by standard methods, either directly by elimination or by iteration. The latter method is employed below.

2. THE MAGNETOSTATIC POTENTIAL

The magnetic induction vector **B** produced by a steady electric current I satisfies Ampère's law,[1]

$$\oint_C \mathbf{B} \cdot d\mathbf{l} = \mu_0 I, \qquad (1)$$

where C is a contour enclosing the conductor carrying I. Employing Stokes's integral theorem, this equation can be written as

$$\operatorname{curl} \mathbf{B} = \mu_0 \mathbf{J}, \tag{2}$$

where $\mathbf{J}$ is the current density (Am^{-2}). The other basic property of $\mathbf{B}$ is that it forms closed loops, i.e. it satisfies

$$\operatorname{div} \mathbf{B} = 0. \tag{3}$$

A small current loop produces a field $\mathbf{B}$ which resembles the electric field near an electric dipole, and consequently a magnetic dipole moment can be identified with the loop. A magnetic material may be thought of as containing a large number of elementary loops, giving rise to a dipole moment per unit volume, $\mathbf{M}$, known as the magnetization. The magnetization contributes to $\mathbf{B}$ and it can be shown[1] quite generally that equation (2) is replaced by

$$\operatorname{curl} \mathbf{B} = \mu_0 \mathbf{J} + \mu_0 \operatorname{curl} \mathbf{M}, \tag{4}$$

where $\mathbf{J}$ is the real current density. It is convenient to define a magnetic field $\mathbf{H}$ by

$$\mathbf{B} = \mu_0(\mathbf{H} + \mathbf{M}), \tag{5}$$

and from equation (4) $\mathbf{H}$ satisfies

$$\operatorname{curl} \mathbf{H} = \mathbf{J}, \tag{6}$$

There are usually no true currents in a permanent magnet so that (6) reduces to

$$\operatorname{curl} \mathbf{H} = 0, \tag{7}$$

and therefore it is possible to define a scalar magnetic potential ϕ by

$$\mathbf{H} = -\nabla\phi. \tag{8}$$

From equations (3) and (5),

$$\operatorname{div} \mathbf{H} = -\operatorname{div} \mathbf{M}, \tag{9}$$

or in terms of ϕ,

$$\nabla^2\phi = \operatorname{div} \mathbf{M}. \tag{10}$$

This is Poisson's equation for the potential and, by analogy with electrostatics, the term $\operatorname{div} \mathbf{M}$ plays the role of a volume magnetic charge density, and is frequently referred to as the pole density.

3. BOUNDARY CONDITIONS ON INTERFACES

The boundary conditions on the magnetostatic potential at the interface between two media in which true currents are absent can be derived from