Table 4. Data for a single lens

n	c	d
1.0		
	0.1	
1.5		0.2
	−0.1	
1.0		
Finite objects	*YOBJ*	1.0
	DOBJ	10.0
	$Y(1)$	1.0
Infinite objects	$Y(1)$	1.0
	$U(1)$	0.0
Use $\Delta\alpha = 0.05$		

compared with the variation in the position of the Gaussian image plane (and focal length) and of the size of the Gaussian image that are, respectively, the longitudinal and transverse chromatic aberrations.

The simplest way to investigate the chromatic effects is, of course, to recalculate f', $S1$, and TA using different values of n. The range of variation of n over the visible spectrum is of the order of a few per cent and is specified by the so-called V-value, where

$$V = (n_D - 1)/(n_F - n_C)$$

and C and F refer to particular red and blue lines in the spectrum of hydrogen and D to a yellow line in the spectrum of sodium, viz.:

Line	Colour	Wavelength (nm)
C	Red	656.3
D	Yellow	589.3
F	Blue	486.1

To find n_F and n_C from the given values of n_D and V we can assume that the refractive index obeys Cauchy's equation:

$$n = A + B/\lambda^2$$

from which it is easily shown that

$$B = (n_D - 1)\lambda_F^2\lambda_C^2/V(\lambda_C^2 - \lambda_F^2)$$

and

$$A = n_D - B/\lambda_D^2.$$

The values of V for the glasses used in the Tessar lens are given in Table 5, together with the calculated values of A, B, n_F, and n_C. A recalculation of

Table 5. Refractive index constants and values

n_D	V	A	B	n_F	n_C
1.6116	58.54	1.5959	5.4688	1.6190	1.6085
1.6053	38.03	1.5813	8.3315	1.6166	1.6007
1.5123	56.35	1.4986	4.7589	1.5187	1.5096
			$\times 10^{-15}$		

Table 6. Data for a telescope object glass*

n	c (cm^{-1})	d (cm)
1.0		
	0.072516	
1.6203		0.290
	−0.16388	
1.5728		0.590
	0.0079681	
1.0		
	LOGICAL = FALSE	
	$Y(1) = 3.00$	
	$U(1) = -3.00°$	

* From reference 7, page 41.

Table 7. Data for a microscope objective

n	c (cm^{-1})	d (cm)
1.0		
	1.96154	
1.572		0.08
	−2.69564	
1.620		0.025
	0.0	
1.0		
	LOGICAL = TRUE	
	$Y(1) = 0.125$ cm	
	$U(1) = 2.38°$	
	$YOBJ = 0.15$ cm	
	$DOBJ = 3.00$ cm	

the Tessar using these values can be carried out to see how it responds to different colours. A good lens should show a variation in focal length of less than 1 part in 2000.

Most elementary optics texts also discuss the realization of achromatism by combining two lenses made of glasses of high and low V-values respectively, and thus as further simple examples one can try the lenses specified in

Tables 6 and 7. These are both cemented achromatic doublets of the type first made by Dollond in 1759, which Newton concluded in 1668 were impossible to produce. The former is an object glass for a telescope and the latter is an objective for a microscope. It is interesting to note that these lenses have minimum spherical aberration when the external shape of the lens agrees with the conditions given in section 9.1.

9.3 The concave mirror

The simplest optical system of all is a single mirror. We have not so far considered reflecting systems, but they can be taken into account by using the fact that when light is reflected the beam reverses direction and Snell's law may be written as

$$\sin I' = -\sin I,$$

that is equivalent to putting

$$n' = -n \text{ or in air, } n = 1,\ n' = -1.$$

The spherical aberration of a concave mirror may thus be obtained by using the data

$$\begin{aligned} \mathrm{NS} &= 1 \\ \mathrm{N(1)} &= 1.0 \\ \mathrm{N(2)} &= -1.0 \\ \mathrm{D(2)} &= 0.0 \end{aligned}$$

and other parameters C(1), Y(1) as required.

9.4 Other aberrations

The program S1BEND calculates only the amount of spherical aberration $S1$ but, as mentioned in section 8.2, there are other aberrations that may be present, of which the most important are coma, $S2$, and astigmatism, $S3$. Coma appears, as shown in Figure 5c, as a difference in focusing position, on the principal ray, for rays above and below the principal ray, whilst astigmatism is a difference in focus for rays in the meridian plane and those in a plane perpendicular to the meridian. Both aberrations can be calculated if the parameters for the paraxial principal ray are known, in particular if $\bar{A}$ (corresponding to A for the paraxial marginal ray) is known at each surface.

The expressions for wavefront coma $S2$ and astigmatism $S3$ are

$$S2 = \tfrac{1}{2} A\bar{A} y \Delta(u/n)$$

$$S3 = \tfrac{1}{2} \bar{A}^2 y \Delta(u/n).$$

The proofs of these expressions are given by Welford,[11] and they may be incorporated into the program by adding the principal ray quantities $\bar{y}$ and $\bar{u}$ to the main program and to the subroutine PARAXL (as COMMON variables) and calculating at each surface

$$\bar{A} = n\bar{u} + n\bar{y}c$$

and thence $S2$ and $S3$. For any object the initial value of $\bar{y}(1)$ is zero at the entrance pupil, whose position may need calculating, as is done in RAYTRAC. For a finite object $\bar{u}(1) = -$YOBJ/DOBJ, for an infinite object $\bar{u}(1) = u(1)$.

REFERENCES

1. H. A. Buchdahl, *An Introduction to Hamiltonian Optics* (University Press, Cambridge), 3 (1970).
2. M. Born and E. Wolf, *Principles of Optics*, 3rd ed. (Pergamon Press, Oxford), 147 (1965).
3. Born and Wolf, *Principles*, 203, also L. Seidel, *Astr. Nachr.*, **43,** 289, 305, 321 (1856).
4. W. T. Welford, *Aberrations of the Symmetrical Optical System* (Academic Press, London), 111 (1974).
5. H. H. Hopkins, *Wave Theory of Aberrations* (Clarendon Press, Oxford), (1950).
6. Welford, *Aberrations*, 49.
7. E. L. O'Neill, *Introduction to Statistical Optics* (Addison-Wesley, Mass.), 43 (1963).
8. Lord Rayleigh, *Phil. Mag.* **8,** 403 (1879).
9. J. M. Palmer, *Lens Aberration Data* (Hilger, Bristol), 56 *et seq.* (1971).
10. F. A. Jenkins and H. E. White, *Fundamentals of Optics*, 4th edn., (McGraw-Hill, New York), 153 (1976).
11. Welford, *Aberrations*, Ch. 7.

S1BEND: PROGRAM TO CALCULATE SPHERICAL ABERRATION

```
C THIS PROGRAM CALCULATES THE FOCAL-LENGTH AND THE INITIAL WAVEFRONT
C  SPHERICAL ABERRATION S1 OF AN OPTICAL SYSTEM. IT ALSO CALCULATES
C  VIA THE SUBROUTINE ALPHA0 THE INITIAL VALUES OF ALPHA (ALPHA=N*I)
C  AND THEN VARIES ALPHA BY +OR- NALPHA*DALPH0 AND RECALCULATES
C  THE ABERRATION
C AS AN ALTERNATIVE IF NALPHA IS SET EQUAL TO ZERO THEN THE VARIATION
C  OF S1 WITH INCIDENCE HEIGHT Y(1) IS DETERMINED FOR VALUES OF Y(1)
C  THAT DIFFER BY SUCCESSIVE INTERVALS OF DELY,WHERE DELY IS SET EQUAL
C  TO THE VALUE OF DALPH0
C THE DATA SHOULD BE READ IN THE FOLLOWING ORDER:
C          FINITE          (FINITE IS TRUE FOR A FINITE OBJECT,
C                           FINITE IS FALSE FOR AN INFINITE OBJECT)
C          NS              (THE NUMBER OF SURFACES)
C          C(J) J=1,NS     (THE SURFACE CURVATURES)
C          N(J) J=1,NS+1   (THE REFRACTIVE INDICES)
C          D(J) J=2,NS     (THE SURFACE SEPARATIONS)
C          Y(1)            (THE INITIAL INCIDENCE HEIGHT)
C          U(1)            (THE INITIAL INCIDENCE ANGLE,IN DEGREES,
C                            U(1) IS POSITIVE FOR A FINITE OBJECT
C                            U(1) IS NEGATIVE FOR AN INFINITE OBJECT)
C          DALPH0,NALPHA   (DALPH0 IS THE INTERVAL BETWEEN SUCCESSIVE
C                            VALUES OF ALPHA AND 2*NALPHA+1 IS THE
C                            TOTAL NUMBER OF VALUES OF ALPHA)
C THE PROGRAM USES THE FOLLOWING SUBROUTINES:
C          ALPHA0  TO CALCULATE THE INITIAL VALUES OF ALPHA
C          FOCUS   TO CALCULATE THE FOCAL LENGTH AND THE
C                   NEW VALUES OF C(J)
C          ABERRN  TO CALCULATE THE WAVEFRONT SPHERICAL ABERRATION
      LOGICAL FINITE
      REAL N
      COMMON NS,N,C,D,U.Y,DALPHA,NALPHA,ALPHA
      DIMENSION N(20),C(20),D(20),U(20),Y(20),ALPHA(20)
      WRITE(2,303)
      READ(1,100) FINITE
      READ(1,101) NS
      WRITE(2,201) NS
      NSP=NS+1
C READ AND WRITE THE SYSTEM PARAMETERS
      READ(1,102)  (C(J),J=1,NS)
      WRITE(2,209)
      WRITE(2,202) (C(J),J=1,NS)
      READ(1,102)  (N(J),J=1,NSP)
      WRITE(2,210)
      WRITE(2,205) (N(J),J=1,NSP)
      READ(1,102)  (D(J),J=2,NS)
      WRITE(2,211)
      WRITE(2,205) (D(J),J=2,NS)
      WRITE(2,301)
C READ AND WRITE THE RAY PARAMETERS
      READ(1,102) Y(1)
      READ(1,102) U(1)
      IF (.NOT.FINITE) GOTO 2
      WRITE(2,204) U(1),Y(1)
      GOTO 5
    2 CONTINUE
      WRITE(2,200) U(1),Y(1)
```

S1BEND: PROGRAM TO CALCULATE SPHERICAL ABERRATION

```
    5 CONTINUE
      IF(ABS(Y(1)).LE.0.0001) Y(1)=0.0001
      PI=3.1415926536
      U(1)=U(1)*PI/180.0
C READ AND WRITE THE VARIATION DALPH0 AND THE NUMBER NALPHA
C  OF VARIATIONS ON EACH SIDE OF THE STARTING POINT
      READ(1,103) DALPH0,NALPHA
      IF(NALPHA.EQ.0) GOTO 23
      NALPH2=2*NALPHA+1
      WRITE(2,207) DALPH0,NALPH2
      CALL ALPHA0
       DO 20 NP=1,NALPH2
      NPM=NP-NALPHA-1
      DALPHA=DALPH0*FLOAT(NPM)
      WRITE(2,208) DALPHA
C NOW CHANGE ALL THE VALUES OF ALPHA
         DO 21 J=1,NS
      ALPHA(J)=ALPHA(J)+DALPHA
   21    CONTINUE
      CALL FOCUS
      CALL ABERRN
C NOW CHANGE ALPHA BACK AGAIN
         DO 22 J=1,NS
      ALPHA(J)=ALPHA(J)-DALPHA
   22    CONTINUE
   20  CONTINUE
      GOTO 29
C IF NALPHA=0 CALCULATE THE VARIATION OF S1 WITH Y(1)
   23 IF(Y(1).LT.0.0) Y(1)=-Y(1)
C STORE THE STARTING VALUE OF Y(1)
      Y1=Y(1)
      CALL FOCUS
      WRITE(2,206)
      DELY=DALPH0
      Y(1)=Y(1)+DELY
   24 Y(1)=Y(1)-DELY
      IF(ABS(Y(1)).LT.0.0001) GOTO 25
      IF(Y(1).LT.0.0) GOTO 27
      CALL ABERRN
      GOTO 24
   25 Y(1)=0.0
      CALL ABERRN
   26 Y(1)=Y(1)-DELY
      CALL ABERRN
      IF(ABS(Y(1)+Y1).LT.0.0001) GOTO 29
      GOTO 26
   27 Y(1)=Y(1)+DELY
      Y(1)=-Y(1)
      CALL ABERRN
   28 Y(1)=Y(1)-DELY
      CALL ABERRN
      IF(ABS(Y(1)+Y1).LT.0.0001) GOTO 29
      GOTO 28
   29 CONTINUE
      STOP
  100 FORMAT(L5)
```

```
SIBEND: PROGRAM TO CALCULATE SPHERICAL ABERRATION

 101 FORMAT(I2)
 102 FORMAT(5F10.0)
 103 FORMAT(F8.0,I2)
 200 FORMAT(23H FIELD ANGLE IN DEGREES,F16.4/1X,16HINCIDENCE HEIGHT,F22
    $.4//)
 201 FORMAT(1X,19H NUMBER OF SURFACES,8X,I2/)
 202 FORMAT(E16.5,4E14.5)
 204 FORMAT(27H INCIDENCE ANGLE IN DEGREES,F12.4/17H INCIDENCE HEIGHT
    $,F22.4/)
 205 FORMAT(F12.5,4F14.5)
 206 FORMAT(6X,8HINCIDENT,8X,9HSPHERICAL/6X,8H HEIGHT ,8X,10HABERRATION
    $/)
 207 FORMAT(19H INTERVALS OF ALPHA,F20.4/26H NUMBER OF VALUES OF ALPHA,
    $I8///)
 208 FORMAT(8X,7HDALPHA=,F8.4/)
 209 FORMAT(1X,23H CURVATURES OF SURFACES)
 210 FORMAT(1X,19H REFRACTIVE INDICES)
 211 FORMAT(1X,24H SEPARATIONS OF SURFACES)
 301 FORMAT(1X/)
 303 FORMAT(1X///)
     END
           SUBROUTINE ALPHA0
C THIS SUBROUTINE SETS UP THE INITIAL VALUES OF ALPHA
     REAL K,N
     COMMON NS,N,C,D,U,Y,DALPHA,NALPHA,ALPHA
     DIMENSION N(20),C(20),D(20),U(20),Y(20),ALPHA(20),UF(20),YF(20)
     UF(1)=0.0
     YF(1)=Y(1)
     J=1
  30 ALPHA(J)=YF(J)*C(J)
     IF (J.EQ.NS) GOTO 31
     K=(N(J+1)-N(J))*C(J)
     UF(J+1)=(N(J)*UF(J)-YF(J)*K)/N(J+1)
     YF(J+1)=YF(J)+D(J+1)*UF(J+1)
     J=J+1
     IF (J.LE.NS) GOTO 30
  31 CONTINUE
     RETURN
     END
           SUBROUTINE FOCUS
C THIS SUBROUTINE CALCULATES THE FOCAL LENGTH FL AND
C  THE NEW VALUES OF C(J)
     REAL K,N
     COMMON NS,N,C,D,U,Y,DALPHA,NALPHA,ALPHA
     DIMENSION N(20),C(20),D(20),U(20),Y(20),ALPHA(20)
     IF(NALPHA.EQ.0) GOTO 41
     C(1)=ALPHA(1)/Y(1)
  41 K=(N(2)-N(1))*C(1)
     U(2)=-Y(1)*K/N(2)
       DO 40 J=2,NS
     Y(J)=Y(J-1)+D(J)*U(J)
     IF(NALPHA.NE.0)  C(J)=ALPHA(J)/Y(J)
     K=(N(J+1)-N(J))*C(J)
     U(J+1)=(N(J)*U(J)-Y(J)*K)/N(J+1)
  40   CONTINUE
     FL=-Y(1)/U(NS+1)
```

```
      BFL=-Y(NS)/U(NS+1)
      WRITE(2,400) FL
      WRITE(2,401) BFL
  400 FORMAT(3X,13H FOCAL LENGTH,F22.5)
  401 FORMAT(3X,18H BACK FOCAL LENGTH,F17.5/)
      RETURN
      END
            SUBROUTINE ABERRN
C THIS SUBROUTINE PERFORMS A PARAXIAL RAYTRACE AND
C  CALCULATES THE WAVEFRONT SPHERICAL ABERRATION
      REAL K,N,NNP
      COMMON NS,N C,D,U,Y,DALPHA,NALPHA,ALPHA
      DIMENSION N(20),C(20),D(20),U(20),Y(20),ALPHA(20)
      DATA CC/1HC/
      S1=0.0
      J=1
C REFRACT
   50 CONTINUE
      K=(N(J+1)-N(J))*C(J)
      U(J+1)=(N(J)*U(J)-Y(J)*K)/N(J+1)
C CALCULATE THE ABERRATION
      A=N(J)*Y(J)*C(J)+N(J)*U(J)
      NNP=N(J+1)*N(J)
      DELUN=(N(J)*U(J+1)-N(J+1)*U(J))/NNP
      S1=S1-0.125*A*A*Y(J)*DELUN
      IF(J.EQ.NS) GOTO 55
C TRANSFER TO THE NEXT SURFACE
      Y(J+1)=Y(J)+D(J+1)*U(J+1)
      J=J+1
      IF(J.LE.NS) GOTO 50
   55 CONTINUE
      IF(NALPHA.EQ.0) GOTO 57
      WRITE(2,500) (CC,J,C(J),J=1,NS)
      WRITE(2,501)
      GOTO 58
   57 WRITE(2,503) Y(1),S1
      GOTO 59
   58 WRITE(2,502) S1
   59 CONTINUE
      RETURN
  500 FORMAT(3(1H ,A1,I2,1H=,E13.6,2X))
  501 FORMAT(1H /)
  502 FORMAT(3X,21H SPHERICAL ABERRATION,E18.5///)
  503 FORMAT(F13.4,E20.5)
      END
```

TYPICAL RESULTS OF A CALCULATION OF SL AS A FUNCTION OF Y(1)

```
NUMBER OF SURFACES          7

CURVATURES OF SURFACES
  0.61425E 00  -0.36271E-01  -0.28927E 00   0.63221E 00   0.00000E 00
  0.52083E 00  -0.41667E 00
REFRACTIVE INDICES
  1.00000       1.61160       1.00000       1.60530       1.00000
  1.51230       1.61160       1.00000
SEPARATIONS OF SURFACES
  0.35700       0.18900       0.08100       0.32500       0.21700
  0.39600


FIELD ANGLE IN DEGREES          -20.0000
INCIDENCE HEIGHT                  0.9000


  FOCAL LENGTH                  5.07986
  BACK FOCAL LENGTH             4.40715

    INCIDENT       SPHERICAL
    HEIGHT         ABERRATION

     0.9000        -0.37623E-02
     0.8000        -0.26028E-02
     0.7000        -0.16942E-02
     0.6000        -0.77258E-03
     0.5000        -0.11107E-03
     0.4000         0.38084E-03
     0.3000         0.70713E-03
     0.2000         0.87522E-03
     0.1000         0.89604E-03
     0.0000         0.78394E-03
    -0.1000         0.55680E-03
    -0.2000         0.23593E-03
    -0.3000        -0.15388E-03
    -0.4000        -0.58436E-03
    -0.5000        -0.10238E-02
    -0.6000        -0.14369E-02
    -0.7000        -0.17851E-02
    -0.8000        -0.20262E-02
    -0.9000        -0.21146E-02
```

```
C THIS PROGRAM TRACES A FINITE MERIDIAN RAY THROUGH AN
C  OPTICAL SYSTEM AND CALCULATES ITS TRANVERSE ABERRATION
C  ON THE GAUSSIAN IMAGE PLANE AS
C        1) TA    WITH RESPECT TO THE GAUSSIAN IMAGE
C        2) TAPR  WITH RESPECT TO THE PRINCIPAL RAY
C THE PROGRAM USES THE SUBROUTINES:
C       FOCUS   TO FIND THE FOCAL LENGTH FL AND THE BACK FOCAL
C                LENGTH BFL
C       PARAXL  TO FIND,FOR A FINITE OBJECT,THE GAUSSIAN IMAGE
C                PLANE GIP AND THE PARAXIAL IMAGE SIZE YPXL
C                (FOR AN INFINITE OBJECT GIP=BFL AND
C                                        YPXL=FL*FIELD ANGLE)
C       PUPIL   TO FIND THE EXIT PUPIL DISTANCE EXPP TO THE RIGHT
C                OF THE FIRST SURFACE AND THE INCIDENT HEIGHT OF
C                THE PRINCIPAL RAY ON THE FIRST SURFACE
C       TRACE   TO TRACE THE FINITE RAYS
C THE DATA SHOULD BE READ IN THE FOLLOWING ORDER:
C       FINITE         (FINITE IS TRUE FOR A FINITE OBJECT,
C                       FINITE IS FALSE FOR AN INFINITE OBJECT)
C       NS             (THE NUMBER OF SURFACES)
C       C(J) J=1,NS    (THE SURFACE CURVATURES)
C       N(J) J=1,NS+1  (THE REFRACTIVES INDICES)
C       D(J) J=2,NS    (THE SURFACE SEPARATIONS)
C       Y(1)           (THE MAXIMUM  INCIDENCE HEIGHT ON THE
C                        FIRST SURFACE)
C       U(1)           (THE FIELD ANGLE IN DEGREES-INCLUDE FOR AN
C                        INFINITE OBJECT-OMIT FOR A FINITE OBJECT)
C       YOBJ,DOBJ      (THE OBJECT HEIGHT AND ITS DISTANCE
C                        TO THE LEFT OF THE FIRST SURFACE
C                        -OMIT FOR AN INFINITE OBJECT)
C       NR             (THE NUMBER OF RAYS TO BE TRACED)
C       DAP,JP         (THE DISTANCE DAP OF THE APERTURE STOP
C                        TO THE RIGHT OF SURFACE NUMBER JP)
      REAL N,K
      COMMON NS,N,C,D,U,Y,FL,BFL,GIP,YGIP,YPXL,YOBJ,DOBJ,EXPP
      DIMENSION N(21),C(20),D(19),U(21),Y(20)
      LOGICAL FINITE
      PI=3.1415926536
      READ(1,100) FINITE
      READ(1,101)  NS
      WRITE(2,201) NS
C READ AND WRITE THE SYSTEM PARAMETERS
      NSP=NS+1
      READ(1,102)  (C(J),J=1,NS)
      READ(1,102)  (N(J),J=1,NSP)
      READ(1,102)  (D(J),J=2,NS)
      WRITE(2,202) (C(J),J=1,NS)
      WRITE(2,203) (N(J),J=1,NSP)
      WRITE(2,204) (D(J),J=2,NS)
C READ THE MAXIMUM INCIDENT HEIGHT Y(1) ON THE TANGENT
C  PLANE AT THE FIRST SURFACE
      READ(1,102) Y(1)
      IF (FINITE) GOTO 10
C FOR AN INFINITE OBJECT READ AND WRITE THE FIELD ANGLE U(1) IN DEGREES
      READ(1,102)  U(1)
      WRITE(2,206) U(1)
```

```
      U(1)=U(1)*PI/180.0
      GOTO 20
C FOR A FINITE OBJECT,READ THE OBJECT HEIGHT YOBJ AND
C  ITS DISTANCE DOBJ  FROM THE FIRST SURFACE
   10 READ(1,102) YOBJ,DOBJ
      WRITE(2,205) YOBJ,DOBJ
C READ THE NUMBER NR OF RAYS TO BE TRACED
   20 READ(1,101) NR
      NRM=NR-1
      DELY=2.0*Y(1)/FLOAT(NRM)
C CALCULATE THE FOCAL LENGTH
      CALL FOCUS
      GIP=BFL
      IF(.NOT.FINITE) YPXL=FL*U(1)
      IF(FINITE) CALL PARAXL
C TO FIND THE PRINCIPAL RAY HEIGHT AT THE FIRST SURFACE
C  FIRST STORE THE INITIAL VALUE OF Y(1) AS Y0
      Y0=Y(1)
      CALL PUPIL
      Y(1)=-EXPP*U(1)
      IF(FINITE) Y(1)=YOBJ*EXPP/(DOBJ+EXPP)
      IF(FINITE) U(1)=-YOBJ/(DOBJ+EXPP)
C TRACE THE PRINCIPAL RAY
      CALL TRACE
      YGIPPR=YGIP
C CALCULATE THE TRANSVERSE ABERRATIONS TA AND TAPR FOR
C  VALUES OF Y(1) THAT SCAN THE APERTURE IN NR STEPS STARTING
C  WITH THE INITIAL VALUE OF Y(1)
C FIRST RETRIEVE THE INITIAL VALUE OF Y(1) FROM Y0
      Y(1)=Y0
      WRITE(2,301)
      Y(1)=Y(1)+DELY
        DO 25 I=1,NR
      Y(1)=Y(1)-DELY
      IF(FINITE) U(1)=ATAN((Y(1)-YOBJ)/DOBJ)*PI/180.0
      CALL TRACE
      TA=YGIP-YPXL
      TAPR=YGIP-YGIPPR
      WRITE(2,207) Y(1),TA,TAPR
   25   CONTINUE
      STOP
  100 FORMAT(L5)
  101 FORMAT(I2)
  102 FORMAT(5F10.0)
  201 FORMAT(1H ///6X,18HNUMBER OF SURFACES,I5/)
  202 FORMAT(11X,10HCURVATURES/5(5F12.5/))
  203 FORMAT(11X,18HREFRACTIVE INDICES/5(5F12.5/))
  204 FORMAT(11X,11HSEPARATIONS/5(5F12.5/))
  205 FORMAT(1H /6X,13HOBJECT HEIGHT,F20.3/6X,15HOBJECT DISTANCE,F18.3/)
  206 FORMAT(1H /6X,11HFIELD ANGLE,F22.3,9H  DEGREES/)
  207 FORMAT(1X,F14.4,E19.4,E18.4)
  301 FORMAT(1H /8X,8HINCIDENT,8X,10HTRANSVERSE,8X,10HT.A.  FROM/8X,
     *8H HEIGHT ,8X,10HABERRATION,6X,14H PRINCIPAL RAY/)
      END
          SUBROUTINE PARAXL
C THIS SUBROUTINE TRACES A PARAXIAL RAY FROM AN AXIAL POINT
```

RAYTRC: PROGRAM TO PERFORM A FINITE RAY TRACE

```
C  ON A FINITE OBJECT AND CALCULATES THE POSITION GIP AND
C  THE SIZE YPXL OF THE GAUSSIAN IMAGE
      REAL K,N
      COMMON NS,N,C,D,U,Y,FL,BFL,GIP,YGIP,YPXL,YOBJ,DOBJ
      DIMENSION N(21),C(20),D(19),U(21),Y(20)
      U(1)=Y(1)/DOBJ
         DO 30 J=1,NS
      K=(N(J+1)-N(J))*C(J)
      U(J+1)=(N(J)*U(J)-Y(J)*K)/N(J+1)
      IF (J.EQ.NS) GOTO 35
      Y(J+1)=Y(J)+D(J+1)*U(J+1)
   30    CONTINUE
   35 GIP=-Y(NS)/U(NS+1)
      YPXL=-(GIP-BFL)*YOBJ/FL
      RETURN
      END
           SUBROUTINE FOCUS
C THIS SUBROUTINE CALCULATES THE FOCAL LENGTH FL
C  AND THE BACK FOCAL LENGTH BFL
      REAL K,N
      COMMON NS,N,C,D,U,Y,FL,BFL
      DIMENSION  C(20),N(21).D(19),U(21),Y(20)
      K=(N(2)-N(1))*C(1)
      IF(Y(1).EQ.0.0) Y(1)=1.0
      U(2)=-Y(1)*K/N(2)
       DO 40 J=2,NS
      Y(J)=Y(J-1)+D(J)*U(J)
      K=(N(J+1)-N(J))*C(J)
      U(J+1)=(N(J)*U(J)-Y(J)*K)/N(J+1)
   40   CONTINUE
      FL=-Y(1)/U(NS+1)
      BFL=-Y(NS)/U(NS+1)
      WRITE(2,401) FL
      WRITE(2,402) BFL
  401 FORMAT(1H0,5X,12HFOCAL LENGTH,F22.4)
  402 FORMAT(6X,17HBACK FOCAL LENGTH,F17.4/)
      RETURN
      END
           SUBROUTINE PUPIL
C THIS SUBROUTINE CALCULATES THE POSITION OF THE EXIT PUPIL
C  FROM THE KNOWN POSITION OF THE APERTURE STOP
      REAL KP,NP,N
      COMMON NS,N,C,D,U,Y,FL,BFL,GIP,YGIP,YPXL,YOBJ,DOBJ,EXPP
      DIMENSTON C(20),N(21),D(19),U(21),Y(20),CP(20),NP(21),DP(20),
     *UP(21),YP(20)
C READ AND WRITE THE DISTANCE DAP OF THE APERTURE STOP
C  TO THE RIGHT OF SURFACE NUMBER JP
      READ(1,103) DAP,JP
      WRITE(2,209) DAP,JP
C SET UP VARIABLES FOR A REVERSE RAY TRACE FROM THE STOP
C  NOTE THAT THE SIGNS OF THE CURVATURES ARE REVERSED
      M=0
   50 M=M+1
      JP2M=JP+2-M
      NP(M)=N(JP2M)
      IF(M.EQ.JP+1) GOTO 55
```

RAYTRC: PROGRAM TO PERFORM A FINITE RAY TRACE

```
      JP1M=JP+1-M
      CP(M)=-C(JP1M)
      IF(M.EQ.JP) GOTO 50
      DP(M+1)=D(JP1M)
      GOTO 50
   55 CONTINUE
C PERFORM A PARAXIAL RAYTRACE TO FIND THE EXIT PUPIL
      IF (ABS(DAP).GE.1.0E-4) GOTO 60
      UP(1)=0.1
      YP(1)=0.0
      GOTO 65
   60 YP(1)=0.1
      UP(1)=YP(1)/DAP
   65 CONTINUE
       DO 70 J=1,JP
      KP=(NP(J+1)-NP(J))*CP(J)
      UP(J+1)=(NP(J)*UP(J)-YP(J)*KP)/NP(J+1)
      IF(J.EQ.JP) GOTO 75
      YP(J+1)=YP(J)+DP(J+1)*UP(J+1)
   70  CONTINUE
   75 EXPP=YP(JP)/UP(JP+1)
C EXPP IS THE EXIT PUPIL PLANE
  103 FORMAT(F8.0,I2)
  209 FORMAT(6X,22HAPERTURE STOP DISTANCE,F8.4,22H TO THE RIGHT OF SURFA
     *,6HCE NO.,I2/)
      RETURN
      END
            SUBROUTINE TRACE
      REAL N,K
      COMMON NS,N,C,D,U,Y,FL,BFL,GIP,YGIP,YPXL,YOBJ,DOBJ
      DIMENSION N(21),C(20),D(19),U(21),Y(20)
C SET UP THE INITIAL VALUES AT THE FIRST SURFACE
      DD=0.0
      Z=0.0
      UU=U(1)
      YY=Y(1)
C DEFINE DIRECTION COSINES DCM AND DCN
      DCM=SIN(UU)
      DCN=COS(UU)
      J=1
C SET UP DUMMY VARIABLES FOR THE LOOP
   85 CC=C(J)
      RN=N(J)
      RNP=N(J+1)
C          TRANSFER
      YY=YY+(DD-Z)*DCM/DCN
      F=CC*YY*YY
      G=DCN-CC*YY*DCM
      COSISQ=G*G-CC*F
      IF(ABS(Y(1)).LE.1.0E-8.AND.ABS(DCM).LE.1.0E-8) GOTO 86
      IF(COSISQ.GE.0.0.AND.COSISQ.LE.1.0) GOTO 87
      YPXL=0.0
      YGIP=9999.0
      WRITE(2,220) J,Y(1)
      GOTO 80
   86 YPXL=0.0
```

RAYTRC: PROGRAM TO PERFORM A FINITE RAY TRACE

```
      YGIP=0.0
      GOTO 80
   87 COSI=SQRT(COSISQ)
      IF(F.EQ.0.0) D0=0.0
      IF(F.EQ.0.0) GOTO 88
      D0=F/(G+COSI)
   88 YY=YY+D0*DCM
      Z=D0*DCN
C        REFRACT
      SINI=SQRT(1.0-COSI*COSI)
      SINIP=RN*SINI/RNP
      IF(SINIP.LE.1.0) GOTO 90
      YPXL=0.0
      YGIP=9999.0
      WRITE(2,221) J,Y(1)
      GOTO 80
   90 COSIP=SQRT(1.0-SINIP*SINIP)
      K=RNP*COSIP-RN*COSI
      DCM=(RN*DCM-YY*K*CC)/RNP
      DCN=(RN*DCN-Z*K*CC+K)/RNP
      IF(J.EQ.NS) GOTO 95
      J=J+1
      DD=D(J)
      GOTO 85
   95 YGIP=YY+(GIP-Z)*DCM/DCN
   80 RETURN
  220 FORMAT(/1X,38HCOSISQ IS OUT OF BOUNDS AT SURFACE NO.,I2,4X,5HY(1)=
     *,F8.4)
  221 FORMAT(/1X,38HCRITICAL ANGLE EXCEEDED AT SURFACE NO.,I2,4X,5HY(1)=
     *,F8.4)
      END
```

TYPICAL RESULTS FROM RAYTRC

NUMBER OF SURFACES 7

CURVATURES

0.61425	-0.03627	-0.28927	0.63221	0.00000
0.52083	-0.41667			

REFRACTIVE INDICES

1.00000	1.61160	1.00000	1.60530	1.00000
1.51230	1.61160	1.00000		

SEPARATIONS

0.35700	0.18900	0.08100	0.32500	0.21700
0.39600				

FIELD ANGLE -20.000 DEGREES

FOCAL LENGTH 5.0799
BACK FOCAL LENGTH 4.4072

APERTURE STOP DISTANCE 0.1000 TO THE RIGHT OF SURFACE NO. 4

INCIDENT HEIGHT	TRANSVERSE ABERRATION	T.A. FROM PRINCIPAL RAY
0.9000	-0.3759E-01	0.2549E-01
0.8000	-0.5342E-01	0.9660E-02
0.7000	-0.6067E-01	0.2415E-02
0.6000	-0.6361E-01	-0.5249E-03
0.5000	-0.6441E-01	-0.1333E-02
0.4000	-0.6421E-01	-0.1126E-02
0.3000	-0.6354E-01	-0.4608E-03
0.2000	-0.6266E-01	0.4163E-03
0.1000	-0.6166E-01	0.1423E-02
-0.0000	-0.6053E-01	0.2551E-02
-0.1000	-0.5929E-01	0.3787E-02
-0.2000	-0.5805E-01	0.5028E-02
-0.3000	-0.5714E-01	0.5936E-02
-0.4000	-0.5741E-01	0.5673E-02
-0.5000	-0.6078E-01	0.2301E-02
-0.6000	-0.7179E-01	-0.8710E-02
-0.7000	-0.1018E 00	-0.3877E-01
-0.8000	-0.1839E 00	-0.1208E 00

CRITICAL ANGLE EXCEEDED AT SURFACE NO. 4 Y(1)= -0.9000

-0.9000	0.9999E 04	0.1000E 05

TYPICAL RESULTS FROM RAYTRC

```
  NUMBER OF SURFACES    7

       CURVATURES
 0.61425    -0.03627    -0.28927     0.63221     0.00000
 0.52083    -0.41667
       REFRACTIVE INDICES
 1.00000     1.61160     1.00000     1.60530     1.00000
 1.51230     1.61160     1.00000
       SEPARATIONS
 0.35700     0.18900     0.08100     0.32500     0.21700
 0.39600

  FIELD ANGLE                  0.000  DEGREES

  FOCAL LENGTH                 5.0799
  BACK FOCAL LENGTH            4.4072

  APERTURE STOP DISTANCE  0.1000 TO THE RIGHT OF SURFACE NO. 4
```

INCIDENT HEIGHT	TRANSVERSE ABERRATION	T.A. FROM PRINCIPAL RAY
0.9000	0.5122E-01	0.5122E-01
0.8000	0.1705E-01	0.1705E-01
0.7000	0.3689E-02	0.3689E-02
0.6000	-0.8162E-03	-0.8162E-03
0.5000	-0.1685E-02	-0.1685E-02
0.4000	-0.1283E-02	-0.1283E-02
0.3000	-0.6604E-03	-0.6604E-03
0.2000	-0.2184E-03	-0.2184E-03
0.1000	-0.2888E-04	-0.2888E-04
-0.0000	0.0000E 00	0.0000E 00
-0.1000	0.2888E-04	0.2888E-04
-0.2000	0.2184E-03	0.2184E-03
-0.3000	0.6604E-03	0.6604E-03
-0.4000	0.1283E-02	0.1283E-02
-0.5000	0.1685E-02	0.1685E-02
-0.6000	0.8162E-03	0.8162E-03
-0.7000	-0.3689E-02	-0.3689E-02
-0.8000	-0.1705E-01	-0.1705E-01
-0.9000	-0.5122E-01	-0.5122E-01

Physics Programs
Edited by A. D. Boardman

CHAPTER 2

Attenuated Total Reflection Analysis of Surface Polaritons

G. C. AERS and A. D. BOARDMAN

1. INTRODUCTION

The internal degrees of freedom of a medium are, generally, excited by the passage through it of an electromagnetic wave. In fact, because of the medium, the electromagnetic wave becomes a new type of wave in which the original electromagnetic field is modified by the induced polarization of the medium. This new coupled mode of excitation is known as a polariton,[1] the exact nature of which is further specified according to which elementary excitation is involved. For example, a photon coupled to the elementary excitation of an electron plasma (as in the case of a metal or semiconductor) is called a plasmon-polariton. A photon coupled to the lattice vibrations in a crystal is called a phonon-polariton, and so on.

The study of such coupled modes provides useful information about the characteristic quantities used to describe the medium. One such important quantity is the dielectric tensor function $\boldsymbol{\varepsilon}$ which relates the displacement vector $\mathbf{D}$ in the medium to the electric field $\mathbf{E}$. This relationship may be written, for an isotropic medium, as $\mathbf{D} = \varepsilon_0 \varepsilon \mathbf{E}$, where ε_0 is the permittivity of free space and ε is now a scalar function of frequency, whose structure can be examined experimentally by using optical techniques.

Particularly useful excitations, in this connection, are surface polaritons;[2] that is excitations which propagate along the boundaries of dielectric media and whose associated fields decay exponentially with distance from a boundary in the direction of the normal (see Figure 1 for isotropic media). Surface polaritons also serve as a sensitive probe of the structure of material surfaces since they are so closely associated with them.

One of the most frequently examined excitations is the surface plasmon-polariton which is a TM wave (magnetic field in the plane of the surface and normal to the propagation direction) and the remainder of this chapter will be devoted to this particular case, although the arguments can be, in principle, easily extended to describe other types of polaritons.





Figure 1. Schematic diagram of the variation of the field amplitudes associated with a surface wave propagating at the boundary between two media. The amplitudes have an oscillatory behaviour in the direction of propagation but die away exponentially in a direction normal to the boundary

Figure 1 shows a surface polariton mode that is a surface guided mode, of the TM type, between two semi-infinite isotropic media. It is not too difficult to show that, for plasmon systems such as metals, it is indeed the TM mode that propagates. Conversely, for magnon systems, such as ferromagnetic insulators, it is the TE mode that is a surface mode. The form of a surface wave is $A_1 \exp[i(kx-\omega t)]\exp(-\beta_1 z)$ for $z>0$ and $A_2 \exp[i(kx-\omega t)]\exp(\beta_2 z)$ for $z<0$. The relationship between angular frequency ω and wave number k is usually called the dispersion equation. This name arises because such an equation determines how a wave packet (pulse) will spread out.

Here, β_1 and β_2 are determined by Maxwell's equations under the TM mode assumption $\mathbf{E}=(E_x, 0, E_z)$, $\mathbf{B}=(0, B_y, 0)$. For a medium with a dielectric function ε the relevant components of Maxwell's equations are

$$\begin{aligned} &\frac{\partial E_x}{\partial z}-ikE_z=i\omega B_y, &&\frac{\partial B_y}{\partial z}=\frac{i\omega}{c^2}\varepsilon E_x,\\ &ikEx+\frac{\partial E_z}{\partial z}=0, &&kB_y=\frac{-\omega}{c^2}\varepsilon E_z, \end{aligned} \tag{1}$$

from which it follows that, if $E_x=A_1e^{\pm\beta z}$,

$$E_z=\pm i\frac{k}{\beta}A_1e^{\pm\beta z}, \qquad B_y=\pm i\frac{\omega\varepsilon}{\beta c^2}A_1e^{\pm\beta z}, \tag{2}$$

where

$$\beta^2=k^2-\varepsilon\omega^2/c^2.$$

Hence

$$\beta_1=\left(k^2-\varepsilon_1\frac{\omega^2}{c^2}\right)^{\frac{1}{2}}, \qquad \beta_2=\left(k^2-\varepsilon_2\frac{\omega^2}{c^2}\right)^{\frac{1}{2}}. \tag{3}$$

The boundary conditions at the interface between the two media are that E_x and B_y are continuous. This leads immediately to

$$\frac{\beta_1}{\beta_2}=-\frac{\varepsilon_1}{\varepsilon_2}>0, \tag{4}$$

and

$$k^2=\frac{\omega^2}{c^2}\frac{\varepsilon_1\varepsilon_2}{\varepsilon_1+\varepsilon_2}>0. \tag{5}$$

Equation (5) is the dispersion equation of the surface waves. Equations (4) and (5) also show that the existence of such waves requires ε_1 and ε_2 to be of opposite sign and their sum to be negative. This can be understood as follows: equation (4) can be satisfied only if $\varepsilon_1>0$, $\varepsilon_2<0$, or vice versa; if this is so then $\varepsilon_1\varepsilon_2<0$, hence $\varepsilon_1+\varepsilon_2<0$, for equation (5) to be satisfied.

A medium with a positive dielectric function is often termed a 'passive medium', and commonly used examples of such media include air, vacuum, glass, or similar media whose dielectric functions are normally fairly constant over the frequency range of interest. The dielectric function for the passive medium will be labelled ε_1 for this work and has the value unity for air or vacuum, 2.25 for typical glasses, and 11.683 for silicon, say. A medium with a negative dielectric function is called an 'active' medium and